
Vorwort

Gleich nachdem die Programmiersprache ABAP mit der Einführung von
ABAP Objects zu Release 4.6 einen großen Schritt in Richtung zeitgemäße
Programmierung vollzogen hatte, stellte sich für die Weiterentwicklung
der Sprache bereits die nächste Herausforderung: Um den Anforderun-
gen eines vom Internet bestimmten, internationalen Programmierumfel-
des gerecht zu werden, musste der Web Application Server als Nachfol-
ger der SAP-Basis und mit ihm die Programmiersprache ABAP unicode-
fähig gemacht werden. Diese Anforderung wurde mit Release 6.10 erfüllt.

Mit der Implementierung von Release 6.10 bzw. 6.20 wird jedoch nie-
mand gezwungen, seine Programme anzupassen, solange das SAP-System
nicht auf Unicode umgestellt wird. Unabhängig davon, ob ein Programm
in einem Unicode- oder einem Nicht-Unicode-System eingesetzt werden
soll, bedeutet die Verwendung der im Zusammenhang mit Unicode ein-
geführten Neuerungen dennoch einen wichtigen Schritt in Richtung
sicherer und fehlerfreier Programmierung.

Daher war vor allem die Einführung von Unicode ein wichtiger Beweg-
grund für dieses Buch. Die im System vorhandene ABAP-Schlüsselwort-
dokumentation wuchs mit der Sprache, indem Neues hinzugefügt und
Altes ungeändert belassen wurde. Es gibt dort beispielsweise ein Teilge-
biet »ABAP Objects« und seit Release 6.10 eben auch ein Teilgebiet
»Unicode«, während die Anweisungen des klassischen Reporting nach
wie vor so beschrieben sind, als wäre jedes ABAP-Programm ein Report,
der mit einer logischen Datenbank verknüpft ist. Da diese Art der Doku-
mentation als Direkthilfe zwar vertretbar, für eine einheitliche Gesamtbe-
schreibung aber nicht mehr ausreichend ist, haben wir die Gelegenheit
ergriffen, der Einführung in die SAP-Programmierung mit ABAP Objects
eine ABAP-Referenz folgen zu lassen.

Das Ziel, das wir uns gesteckt haben, ist die vollständige Neufassung und
Vereinheitlichung der ABAP-Dokumentation. Dies betrifft nicht nur die
Gliederung und Zusammenfassung der einzelnen Anweisungen, sondern
auch die inhaltliche Aufbereitung und stilistische Anpassung der Beschrei-
bung. Speziell für die Darstellung der Syntaxdiagramme wurde daher eine
neue Sichtweise in Form einer Pseudosyntax eingeführt, die jede Anwei-
sung ausgehend von ihrer Grundform schrittweise in ihre einzelnen
Bestandteile auflöst. Erstmals werden auch die Anforderungen an alle in
einer ABAP-Anweisung vorkommenden Operanden vollständig und in
einer einheitlichen Form beschrieben. Unser Anspruch ist die umfassende

Beschreibung aller im aktuellen Release 6.20 zur Verwendung freige-
gebenen ABAP-Sprachelemente. Dies beinhaltet auch eine detaillierte
Beschreibung aller zwar als obsolet gekennzeichneten, aber nur in ABAP
Objects verbotenen Sprachelemente. Die ABAP-Referenz soll den Leser
in die Lage versetzen, jedes ABAP-Programm zu analysieren und neue
Programme zu schreiben. Sie kann und will zwar kein Best-Practices-Buch
sein, wir hoffen aber, dass Sinn und Zweck mancher Anweisung durch die
genaue Beschreibung klarer wird, als es bisher der Fall war.

Es ist leichter gesagt als getan, eine Sprache mit etwa 500 wesentlichen
Sprachelementen durchgehend auf gleich hohem Niveau und in gleichem
Detaillierungsgrad zu beschreiben, zumal ABAP seit zwanzig Jahren stän-
dig weiterentwickelt wird und aus Kompatibilitätsgründen keine Entwick-
lung jemals wieder zurückgenommen wurde. Dies erklärt auch die wie-
derholten Verschiebungen des Erscheinungstermins dieses Buches und
wir danken allen, die so lange gewartet haben. Als Entschädigung halten
Sie jetzt aber ein Buch in Händen, das das aktuellste Release der Sprache
ABAP beschreibt und lange gültig bleiben wird. Da wir jedoch trotz sorg-
fältiger Durchsicht aller Kapitel nicht ausschließen können, dass dieses
Buch noch kleinere Unstimmigkeiten enthält, freuen wir uns auf alle dies-
bezüglichen Anregungen und Hinweise. Sie können sie auf den Internet-
seiten von SAP PRESS unter www.sap-press.de einreichen.

Ohne die mittelbare und unmittelbare Hilfe vieler Personen bei der Erstel-
lung und Prüfung des Manuskripts wäre das Buch in der vorliegenden
Form nicht zustande gekommen. An dieser Stelle gilt der Dank den
Kollegen Masoud Aghadavoodi, Thomas Bareiss, Adrian Goerler, Christian
Jendel, Gerd Kluger, Björn Mielenhausen, Andreas Simon Schmitt und
Christoph Stöck. Erhardt Vortanz danken wir für seine Mitarbeit bei der
Erstellung des Manuskripts. Dem Development Manager der Gruppe
Business Programming Languages, Andreas Blumenthal, danken wir, dass
er uns dieses Projekt überhaupt ermöglichte und weitgehend freie Hand
in der Gestaltung ließ. Dem unermüdlichen Michael Demuth ist es zu ver-
danken, dass diesem Buch wieder zwei CDs mit einem SAP-System, dieses
Mal in Form eines Web Application Server 6.10, beiliegen. Dank gilt
schließlich auch allen Mitarbeitern von Galileo Press – insbesondere Iris
Warkus und Florian Zimniak – für die gute Zusammenarbeit bei der Kor-
rektur des Manuskripts und dafür, dass sie die Hoffnung nie aufgaben,
dass wir unser Manuskript irgendwann überhaupt noch einmal einreichen
würden.

Horst Keller dankt ganz besonders seiner Frau Ute, die ihn in den letzten
Monaten fast nur noch über dem aufgeklappten Laptop zu Gesicht
bekam, für ihre Geduld und ihr Verständnis dafür, dass ein großer Anteil
der ohnehin spärlichen gemeinsamen Freizeit schon wieder für ein Buch-
projekt geopfert wurde. 

Wir wünschen nun allen Leserinnen und Lesern viel Spaß beim Durchstö-
bern dieser ABAP-Referenz.

Walldorf, im Juli 2002
Horst Keller, Joachim Jacobitz

1 Einführung und Übersicht

1.1 Ziel des Buches

Das vorliegende Buch enthält eine vollständige Beschreibung der Pro-
grammiersprache ABAP, die seit 1982 bei der SAP AG entwickelt und in
den meisten SAP-Anwendungsprogrammen eingesetzt wird. Es behan-
delt Aufbau und Wirkungsweise aller zur Verwendung freigegebenen
ABAP-Anweisungen in detaillierter Form und ist damit ein umfassendes
Nachschlagewerk für alle Release-Stände bis einschließlich 6.20. 

Das Buch ist als Arbeitsbuch für den täglichen Einsatz jedes ABAP-Ent-
wicklers gedacht. Für eine Einführung in die Programmiersprache ABAP
verweisen wir auf das Buch »ABAP Objects – Einführung in die SAP-Pro-
grammierung« von Horst Keller und Sascha Krüger, das ebenfalls bei SAP
PRESS erschienen ist.

1.2 Beschriebene Releases

Wir beschreiben den Stand der Sprache ABAP zu Release 6.20. Neuerun-
gen, die sich von Release 4.6 zu Release 6.10 und 6.20 ergeben haben,
sind in der Randspalte mit Ikonen gekennzeichnet, so dass das Buch
uneingeschränkt auch für Release 4.6 verwendbar ist. Folgende wesentli-
che Änderungen haben sich von Release 4.6 nach 6.10 ergeben:

� ABAP-Programme können unicode-fähig gemacht werden.

� Es wurde ein neues, klassenbasiertes Ausnahmekonzept eingeführt.

� Es gibt neue Sprachelemente für die dynamische Programmierung, wie
z.B. neue generische Datentypen, typisierte Datenreferenzen und die
Möglichkeit, Open-SQL-Anweisungen weitgehend dynamisch anzuge-
ben.

Selbstverständlich werden mit der Beschreibung von Release 6.20 bzw.
Release 4.6 auch alle vorhergehenden Releases abgedeckt.1 Da Release
4.6 aber der inzwischen verbreitete Standard ist, wird bei den Sprach-
elementen, die vor Release 6.10 eingeführt wurden, nicht vermerkt, ab
welchem Release sie gültig sind. Hierfür sei auf die ABAP-Schlüsselwort-
dokumentation verwiesen, in der dies unter dem Knoten ABAP • Release-
abhängige Änderungen seit Release 3.0 dokumentiert ist.

1 Manches betagte Sprachelement wurde sogar bisher noch nie so detailliert wie hier
beschrieben.

1.3 SAP-Basis und Web Application Server
Zu Release 6.10 wurde die SAP-Basis zum Web Application Server weiter-
entwickelt. Der Web Application Server enthält die Funktionalität der
SAP-Basis und damit die ABAP-Laufzeitumgebung. Unter anderem unter-
scheiden folgende Komponenten den Web Application Server von der
alten SAP-Basis:

� Der Web Application Server unterstützt HTTP(S) und kann damit
direkt, d.h. ohne eine Zwischenstufe wie sie beispielsweise der Inter-
net Transaction Server (ITS) darstellt, als Webserver eingesetzt werden. 

� Für die Programmierung eines Webservers wurden Business Server
Pages (BSP) eingeführt. Business Server Pages sind HTML-Seiten, die
ein Server Side Scripting mit ABAP und JavaScript als Script-Sprachen
erlauben. Durch die Einbindung von ABAP als Script-Sprache hat man
in Business Server Pages direkten Zugriff auf den gesamten Sprachum-
fang von ABAP und damit insbesondere über Open SQL Zugriff auf die
Datenbank des Web Application Server.

� Der Web Application Server enthält einen XSLT-Prozessor, der die
Umwandlung von mit XML beschriebenen Daten in ABAP-Datentypen
und umgekehrt ermöglicht.

� Der Web Application Server enthält einen JavaScript-Prozessor, der es
ermöglicht, in JavaScript geschriebene Programme auszuführen. Vari-
ablen des Scripts können an Datenobjekte eines ABAP-Programms
angebunden werden.

In diesem Buch beschreiben wir ausschließlich die Sprache ABAP. Für
weitergehende Informationen zum Web Application Server, insbesondere
zu Business Server Pages, verweisen wir auf das Buch »SAP Web Applica-
tion Server – Entwicklung von Web-Anwendungen« von Frédéric Heine-
mann und Christian Rau, das ebenfalls bei SAP PRESS erscheint. 

1.4 ABAP und Unicode
Vor Release 6.10 wurden von SAP unterschiedliche Codes für die Darstel-
lung von Zeichen verschiedener Schriften verwendet, wie zum Beispiel
ASCII, EBCDIC oder Double-Byte-Codepages:

� ASCII (American Standard Code for Information Interchange) verschlüs-
selt jedes Zeichen durch ein Byte. Damit lassen sich höchstens 256 Zei-
chen darstellen.2 Gebräuchliche Codepages sind zum Beispiel
ISO88591 für westeuropäische oder ISO88595 für kyrillische Schriften.

2 Genaugenommen verschlüsselt Standard-ASCII ein Zeichen nur durch 7 Bit und
kann daher nur 128 Zeichen abbilden.

� EBCDIC (Extended Binary Coded Decimal Interchange) verschlüsselt
ebenfalls jedes Zeichen durch ein Byte, womit sich wiederum 256 Zei-
chen abbilden lassen. EBCDIC 0697/0500 ist beispielsweise ein IBM-
Format, das zum Beispiel auf der Plattform AS400 für westeuropäische
Schriften verwendet wird.

� Double-Byte-Codepages benötigen 1 bis 2 Byte je Zeichen. Dadurch
lassen sich 65536 Zeichen darstellen, wobei in der Regel nur 10000
bis 15000 Zeichen belegt sind. Als Codepages werden zum Beispiel
SJIS für die japanische und BIG5 für die traditionelle chinesische Schrift
verwendet.

Mit diesen Zeichensätzen können in einem SAP-System alle Sprachen ein-
zeln abgedeckt werden. Schwierigkeiten ergeben sich, wenn in einem
zentralen System Texte aus verschiedenen inkompatiblen Zeichensätzen
gemischt werden sollen. Auch der Austausch von Daten zwischen Syste-
men mit inkompatiblen Zeichensätzen kann zu Problemen führen.

Die Lösung dieses Problems ist die Verwendung eines Zeichensatzes, der
alle Zeichen auf einmal umfasst; dies wird durch Unicode (ISO/IEC
10646) verwirklicht. Für den Unicode-Zeichensatz gibt es verschiedene
Unicode-Zeichendarstellungen, in denen ein Zeichen zwischen ein und
vier Bytes belegen kann. 

Der Web Application Server unterstützt ab Release 6.10 sowohl Nicht-
Unicode- als auch Unicode-Systeme. Nicht-Unicode-Systeme sind her-
kömmliche SAP-Systeme, bei denen in der Regel ein Zeichen durch ein
Byte repräsentiert wird. Unicode-Systeme sind SAP-Systeme, die auf
einer Unicode-Zeichendarstellung basieren und denen ein entsprechen-
des Betriebssystem samt Datenbank zugrunde liegt.

Vor Release 6.10 gingen viele ABAP-Programmiertechniken davon aus,
dass ein Zeichen einem Byte entspricht. Bevor ein System auf Unicode
umgestellt wird, müssen ABAP-Programme deshalb überall dort geändert
werden, wo eine explizite oder implizite Annahme über die interne Länge
eines Zeichens gemacht wird. 

ABAP unterstützt diese Umstellung durch neue Syntaxregeln und neue
Sprachkonstrukte, wobei größter Wert darauf gelegt wurde, so viel vor-
handenen Quelltext wie nur möglich zu erhalten. Als Vorbereitung auf
eine Umstellung auf Unicode – aber auch unabhängig davon, ob ein Sys-
tem tatsächlich auf Unicode umgestellt werden soll – kann in den Pro-
grammeigenschaften das Ankreuzfeld Unicodeprüfungen aktiv markiert
werden. Die Transaktion UCCHECK unterstützt das Einschalten dieser

Prüfung für vorhandene Programme. Falls diese Eigenschaft gesetzt ist,
wird das Programm als Unicode-Programm bezeichnet. In einem Uni-
code-Programm wird eine andere, strengere Syntaxprüfung durchgeführt
als in Nicht-Unicode-Programmen. In einigen Fällen müssen Anweisun-
gen auch um neue Zusätze erweitert werden. Das vorliegende Buch
beschreibt die Regeln für Unicode- und Nicht-Unicode-Programme. Ein
syntaktisch korrektes Unicode-Programm läuft in der Regel mit gleicher
Semantik und gleichen Ergebnissen in Unicode- und Nicht-Unicode-Sys-
temen.3 Programme, die in beiden Systemen laufen sollen, sollten daher
auch auf beiden Plattformen getestet werden. 

In einem Unicode-System können nur Unicode-Programme ausgeführt
werden. Vor der Umstellung auf ein Unicode-System muss der Profilpara-
meter ABAP/UNICODE_CHECK auf »ON« gesetzt werden, so dass nur
noch die Ausführung von Unicode-Programmen erlaubt ist. Nicht-Uni-
code-Programme können nur in Nicht-Unicode-Systemen ausgeführt
werden. Alle Sprachkonstrukte, die für Unicode-Programme eingeführt
wurden, können aber auch in Nicht-Unicode-Programmen verwendet
werden. 

Es hat sich herausgestellt, dass Programme, die bisher schon sauber pro-
grammiert wurden, die neuen Unicode-Regeln größtenteils schon auto-
matisch erfüllen und deshalb kaum Änderungsaufwand hervorrufen.
Umgekehrt deuten alle Stellen, an denen größere Änderungen vorzuneh-
men sind, meistens auf einen fehleranfälligen und daher fragwürdigen Pro-
grammierstil hin. Selbst wenn es nicht geplant ist, auf ein Unicode-System
umzustellen, sind Unicode-Programme die besseren Programme, weil sie
besser wartbar und weniger fehleranfällig sind. So wie in ABAP Objects
veraltete und gefährliche Sprachkonstrukte als obsolet erklärt wurden und
nicht mehr verwendet werden dürfen, bieten die Regeln für Unicode-Pro-
gramme erhöhte Sicherheit beim Programmieren, beispielsweise beim
Arbeiten mit Zeichenfeldern und gemischten Strukturen. Dies gilt insbe-
sondere auch bei der Ablage externer Daten wie z.B. über die
Dateischnittstelle, die für die Verwendung in Unicode-Programmen voll-
ständig überarbeitet wurde. Es wird deshalb empfohlen, beim Anlegen
eines neuen Programms dieses immer als Unicode-Programm zu kenn-
zeichnen und ältere Programme schrittweise auf Unicode umzustellen. 

3 Ausnahmen von dieser Regel sind systemnahe Programme, die die Anzahl der Bytes
pro Zeichen abfragen und auswerten.

1.5 Aufbau des Buches

Das Buch ist in Kapitel unterteilt, die thematisch zusammengehörige
Anweisungen beschreiben. Jedes dieser Kapitel enthält eine Übersicht,
die in das Thema einführt. Wenn ein einziges ABAP-Schlüsselwort seman-
tisch unterschiedliche Anweisungen einleitet, wie zum Beispiel GET oder
CALL, haben wir diese nicht in einer Syntaxform zusammengefasst, son-
dern entsprechend der Semantik aufgeteilt und in den thematisch zuge-
hörigen Abschnitten angeordnet.

Abbildung 1.1 zeigt den allgemeinen Aufbau eines ABAP-Programms.
Welche Komponenten im konkreten Fall tatsächlich verwendet werden,
richtet sich nach den Anforderungen an das Programm und nach seinem
Programmtyp.

Abbildung 1.1  Allgemeiner Aufbau eines ABAP-Programms

An der Verwendung der beschriebenen Anweisungen in ABAP-Program-
men orientiert sich auch die Reihenfolge der Kapitel dieses Buches:

� In Kapitel 2 beschreiben wir die allgemeine ABAP-Syntax. Wir definie-
ren Begriffe wie Anweisung, Schlüsselwort, Sprachelement etc. und
gehen insbesondere darauf ein, wie Datenobjekte in Operandenposi-
tionen angegeben werden.

� Jedes ABAP-Programm beginnt mit einer programmeinleitenden Anwei-
sung. Diese beschreiben wir in Kapitel 3.

� Die Funktionalität eines ABAP-Programms ist in Verarbeitungsblöcken
implementiert. Verarbeitungsblöcke werden mit Modularisierungsan-
weisungen definiert. In Verarbeitungsblöcken vom Typ »Prozedur« kön-
nen mit deklarativen Anweisungen lokale Datentypen und Datenob-
jekte angelegt werden. Alle anderen Verarbeitungsblöcke haben keinen
lokalen Datenbereich und deklarative Anweisungen wirken programm-
global. Eine Sonderrolle spielen Methoden. Methoden sind Verarbei-
tungsblöcke, die im Implementierungsteil ihrer Klasse implementiert
werden müssen. Die Reihenfolge von Verarbeitungsblöcken bzw. Im-
plementierungsteilen spielt keinerlei Rolle für die Programmausfüh-
rung, sollte aber die Lesbarkeit eines Programms unterstützen. Die Mo-
dularisierungsanweisungen beschreiben wir in Kapitel 4.

� In jedem ABAP-Programm stehen vordefinierte Datentypen und Daten-
objekte sowie vordefinierte Funktionen zur Verfügung. Diese führen wir
in Kapitel 5 auf.

� Nach der programmeinleitenden Anweisung enthält jedes Programm
einen globalen Deklarationsteil, in dem Definitionen und Deklaratio-
nen vorgenommen werden, die im gesamten Programm gültig und
sichtbar sind. Dies sind zum einen die Deklaration von Datentypen und
Datenobjekten und zum anderen die Definition von Interfaces oder der
Deklarationsteil von Klassen in ABAP Objects. Die Definitionen von
Interfaces und Klassen enthalten die Deklarationen ihrer Komponen-
ten. Die Reihenfolge der einzelnen Definitionen und Deklarationen ist
zwar nicht prinzipiell festgelegt, muss sich aber nach der Tatsache rich-
ten, dass man sich in einer ABAP-Anweisung immer nur auf vorherge-
hende Definitionen und Deklarationen beziehen kann. Beispielsweise
kann sich eine Referenzvariable nur auf eine zuvor definierte Klasse
beziehen und diese wiederum nur ein zuvor definiertes Interface im-
plementieren. Nach der Einleitung einer Prozedur können Datentypen
und Datenobjekte deklariert werden, die innerhalb der Prozedur sicht-
bar sind. Wir beschreiben die deklarativen Anweisungen für Daten-

typen und Datenobjekte in Kapitel 6 und die Anweisungen zur Defini-
tion von Klassen und Interfaces in Kapitel 7. Zur Deklaration von Da-
tentypen gehören in weiterem Sinn auch Typisierungen von Objekten,
deren Datentyp bei der Programmerstellung noch nicht feststeht.
Diese beschreiben wir in Kapitel 8.

� Alle übrigen Anweisungen eines ABAP-Programms sind Implementie-
rungsanweisungen, die immer einem Verarbeitungsblock zugeordnet
werden können. Mit den Implementierungsanweisungen wird die
Funktionalität eines Verarbeitungsblocks implementiert. In den Kapi-
teln 9 bis 40 beschreiben wir die Implementierungsanweisungen
geordnet nach ihrer Funktionalität. Beachten Sie, dass die Funktionali-
tät aller Verarbeitungsblöcke im Wesentlichen mit den gleichen
Anweisungen implementiert wird. Aus Gründen der Abwärtskompa-
tibilität sind in Verarbeitungsblöcken außerhalb von ABAP Objects
einige Syntaxvarianten möglich, die in Methoden verboten sind. In der
Regel sollten diese Varianten prinzipiell nicht mehr verwendet werden.
Wir beschreiben die ABAP-Syntax, wie sie in Methoden gültig ist, und
weisen gegebenenfalls auf die obsoleten Varianten hin. 

� Da ABAP eine historisch gewachsene Sprache ist, enthält es einige
ältere Anweisungen oder Zusätze zu Anweisungen, die durch bessere
Sprachkonstrukte ersetzt wurden, aus Gründen der Abwärtskompa-
tibilität aber nicht abgeschafft werden können. In ABAP Objects, d.h.
bei der Implementierung von Methoden, sind die obsoleten Sprach-
elemente in der Regel syntaktisch verboten. Sie sollten diese Sprach-
elemente in neuen Programmen zwar prinzipiell nicht mehr verwen-
den, werden sie aber in älteren Programmen weiterhin vorfinden. Wir
beschreiben die obsoleten Anweisungen in Kapitel 41. Obsolete
Zusätze zu weiterhin aktuellen Anweisungen beschreiben wir in der
Regel entsprechend gekennzeichnet bei der Anweisung selbst, teil-
weise aber auch in Kapitel 41. 

� Im Anhang werden die Konvertierungsregeln für Zuweisungen, einige
sprachnahe Klassen und Funktionsbausteine und die Ausnahmen der
ABAP-Laufzeitumgebung dargestellt. 

� Die Beschreibung der ABAP-Sprachelemente wird durch ein umfang-
reiches lexikalisch aufgebautes Glossar ergänzt, das wichtige Begriffe
zentral definiert und Verweise auf die zugehörigen Abschnitte enthält.
Ein im Text kursiv gesetzter Ausdruck verweist auf einen Eintrag im
Glossar. 

Hinweis
Begriffe in der männlichen Form wie »Benutzer« oder »Entwickler« wer-
den in diesem Buch lediglich im grammatikalischen Sinn und nur der Ein-
fachheit halber verwendet und sollen keinesfalls Benutzerinnen oder Ent-
wicklerinnen ausschließen.

1.6 Suchmöglichkeiten

Für das Arbeiten mit der ABAP-Referenz bieten sich drei Zugangsmöglich-
keiten an:

1. Suche nach einem Sprachelement
Die häufigste Anwendung dieses Buches wird wahrscheinlich sein, dass
Sie die genaue Bedeutung eines ABAP-Sprachelements nachschlagen
wollen. Hierfür ist der ausführliche zweistufige Index vorgesehen, der
ausschließlich ABAP-Sprachelemente enthält. Zusätze zu Schlüsselwör-
tern werden auf oberster Ebene dargestellt und darunter finden Sie die
Verweise auf die Schlüsselwörter, zu denen die Zusätze gehören.

2. thematische Suche
Wenn Sie sich über alle zu einem bestimmten Thema gehörigen Sprach-
elemente informieren wollen, suchen Sie am besten im Inhaltsver-
zeichnis nach dem gewünschten Thema. In der Übersicht zum Thema
finden Sie eine Liste aller im betreffenden Kapitel beschriebenen ABAP-
Anweisungen.

3. Suche über einen Begriff
Wenn Sie sich über einen Begriff aus dem ABAP-Umfeld und die zuge-
hörigen Sprachelemente informieren wollen, bietet es sich an, den
Begriff im ABAP-Glossar nachzuschlagen. Bei allen Begriffen des Glos-
sars, die in Bezug zu einem Sprachelement stehen, gibt es einen ent-
sprechenden Verweis.

Natürlich können Sie neben der gezielten Suche einfach auch einmal auf
eine kleine Entdeckungsreise durch die ABAP-Referenz gehen.

1.7 Syntaxdiagramme

Tabelle 1.1 zeigt die in den Syntaxdiagrammen dieses Buches verwendete
Notation. Der wesentliche Unterschied zur Darstellung in der ABAP-
Schlüsselwortdokumentation ist die Verwendung von Pseudosyntax, um
Anweisungen mit zahlreichen Varianten in einer einzigen Grundform dar-
stellen zu können. Pseudosyntax wird weiterhin dazu verwendet, um bei
Anweisungen mit vielen Zusätzen diese thematisch zu ordnen.

Darstellung Bedeutung

ABAP ABAP-Wörter werden in Großbuchstaben und in Fettdruck 
dargestellt. 

operand Operanden werden in Kleinbuchstaben dargestellt.

pseudo_syntax Teile von Anweisungen, deren Syntaxdiagramme an späte-
rer Stelle gezeigt werden, sind als Pseudosyntax in Klein-
buchstaben und Fettdruck dargestellt.

. : , () Punkte, Kommata, Doppelpunkte und runde Klammern 
werden normal dargestellt. Sie sind Teil der ABAP-Syntax.

+, -, *, / ... Operatoren werden normal dargestellt. Sie sind Teil der 
ABAP-Syntax.

[] Teile von Anweisungen, die verwendet werden können, 
aber nicht müssen, sind in eckige Klammern gesetzt. Eine 
Aufzählung von Anweisungsteilen in eckigen Klammern 
bedeutet, dass alle oder einzelne Teile verwendet werden 
dürfen. Wenn mindestens ein Teil verwendet werden 
muss, ist dies im Text vermerkt. Eckige Klammern sind 
nicht Teil der ABAP-Syntax.

| Striche zwischen Anweisungsteilen bedeuten, dass nur 
einer der aufgeführten Teile innerhalb einer Anweisung 
verwendet werden darf. Striche sind nicht Teil der ABAP-
Syntax.

{} Geschweifte Klammern fassen zusammengehörige Teile 
von Anweisungen zusammen, beispielsweise rechts oder 
links von Strichen (|). Geschweifte Klammern sind nicht 
Teil der ABAP-Syntax.

... Punkte bedeuten, dass an dieser Stelle andere Teile der 
Anweisung stehen können. 

operandi Operanden mit tiefergestellten Indizes bedeuten, dass an 
dieser Stelle eine Liste von Operanden aufgeführt werden 
kann. Die einzelnen Operanden sind nur durch Leerzei-
chen getrennt.

{ ... operandi ...}
[ ... operandi ...]

Befinden sich Operanden mit Index innerhalb einer 
geschweiften oder eckigen Klammer, bedeutet das, dass 
der gesamte Inhalt der Klammer als Liste mit unterschied-
lichen Operanden aufgeführt werden kann. 

(dobji) Befinden sich Operanden mit Index innerhalb einer runden 
Klammer, bedeutet das, dass in der Klammer eine Liste 
von durch Kommata getrennten Operanden aufgeführt 
werden kann.

Tabelle 1.1  Syntaxkonventionen

Die Syntaxdiagramme dieses Buches zeigen die Sprachelemente in einer
syntaktisch korrekten Reihenfolge. In vielen Anweisungen sind auch
andere Reihenfolgen möglich, werden aber nicht erwähnt. Wir haben in
der Regel die Reihenfolge gewählt, die am besten zur Semantik der
Anweisung passt und mit der sich die verschiedenen Varianten einer
Anweisung einheitlich beschreiben lassen. Es kann deshalb vorkommen,
dass Sie in bestehenden Programmen oder der ABAP-Dokumentation
eine andere Reihenfolge vorfinden.

Hinweis
Die Wahl von Großbuchstaben für ABAP-Worte und von Kleinbuchsta-
ben für Operanden orientiert sich an einer gängigen Pretty-Printer-Ein-
stellung des ABAP Editor, wie sie auch in der ABAP-Dokumentation ver-
wendet wird. Eine andere Einstellung des Pretty Printer, die seit Release
6.10 auswählbar ist, ermöglicht auch die umgekehrte Darstellung. 

1.8 Weitere Informationen zum Buch

Weitere Informationen zum Buch, eventuelle Korrekturlisten, ein Diskus-
sionsforum und Updates finden Sie im Internet unter www.sap-press.de. 

7 Definition von Klassen und Interfaces

7.1 Übersicht 

Dieser Abschnitt beschreibt die Definition von Klassen und Interfaces
sowie ihrer Komponenten. Klassen und Interfaces sind die Grundlage von
ABAP Objects, dem objektorientierten Bestandteil der Sprache ABAP.
Klassen und Interfaces können in ABAP-Programmen folgender Pro-
grammtypen (siehe Tabelle 3.1) definiert werden:

� In einem Class-Pool kann mit dem Werkzeug Class Builder der ABAP
Workbench genaus eine globale Klasse der Klassenbibliothek zur Ver-
wendung in allen übrigen ABAP-Programmen definiert werden. Wei-
terhin können in einem Class-Pool lokale Klassen und Interfaces zur
Verwendung im Class-Pool selbst definiert werden.

� In einem Interface-Pool kann mit dem Werkzeug Class Builder der ABAP
Workbench genau ein globales Interface der Klassenbibliothek zur Ver-
wendung in allen übrigen ABAP-Programmen definiert werden. In
einem Interface-Pool können keine lokalen Klassen und Interfaces
definiert werden.

� In allen übrigen ABAP-Programmen außer Typgruppen können lokale
Klassen und Interfaces zur Verwendung im Programm selbst definiert
werden.

Innerhalb von Klassen und Interfaces können außer den in Kapitel 6
beschriebenen Deklarationen weitere Komponenten deklariert werden,
die wir ebenfalls in diesem Abschnitt beschreiben.

Die Anweisungen zur Deklaration von Klassen und Interfaces sind:

Anweisung Abschnitt

CLASS 7.2

INTERFACE 7.3

METHODS, CLASS-METHODS 7.4.1

EVENTS, CLASS-EVENTS 7.4.2

INTERFACES, ALIASES 7.4.3

7.2 Definition von Klassen

Die vollständige Definition einer Klasse besteht aus einem Deklarations-
teil und einem Implementierungsteil, die beide mit CLASS eingeleitet wer-
den. Im Deklarationsteil werden die Eigenschaften der Klasse festgelegt
und ihre Komponenten deklariert. Im Implementierungsteil werden die
Methoden der Klasse implementiert. Weitere Varianten von CLASS die-
nen dem Bekanntmachen von Klassen in einem Programm.

7.2.1 Deklarationsteil

Definition einer Klasse und Deklaration ihrer Komponenten.

Syntax

Der Anweisungsblock CLASS class DEFINITION – ENDCLASS definiert
eine Klasse class. Zwischen CLASS und ENDCLASS werden die Kompo-
nenten components der Klasse deklariert. Jede Komponente muss hinter
einer der Anweisungen PUBLIC SECTION, PROTECTED SECTION oder
PRIVATE SECTION, und diese müssen in der angegebenen Reihenfolge
aufgeführt werden. Eine Klasse muss nicht alle SECTION-Anweisungen
enthalten.

Mit den Zusätzen der Anweisung CLASS kann eine Klasse global in der
Klassenbibliothek veröffentlicht, eine Vererbungsbeziehung definiert, die
Klasse abstrakt oder final gemacht, die Instanzierbarkeit gesteuert und
anderen Klassen oder Interfaces die Freundschaft angeboten werden.

CLASS

CLASS class DEFINITION [PUBLIC]
                       [INHERITING FROM superclass]
                       [ABSTRACT]
                       [FINAL]
                       [CREATE {PUBLIC|PROTECTED|PRIVATE}]
                       [[GLOBAL] FRIENDS [classi] [ifaci]].
  [PUBLIC SECTION].
    [components]
  [PROTECTED SECTION].
    [components]
  [PRIVATE SECTION].
    [components]
ENDCLASS.

7.2.1.1 Sichtbarkeitsbereiche

Öffentlicher Sichtbarkeitsbereich

Syntax

Diese Anweisung definiert den öffentlichen Sichtbarkeitsbereich der
Klasse class. Alle Komponenten der Klasse, die im Bereich hinter der
Anweisung PUBLIC SECTION deklariert werden, sind von außerhalb der
Klasse, in ihren Unterklassen und in der Klasse selbst ansprechbar.

Geschützter Sichtbarkeitsbereich

Syntax

Diese Anweisung definiert den geschützten Sichtbarkeitsbereich der
Klasse class. Alle Komponenten der Klasse, die im Bereich hinter der
Anweisung PROTECTED SECTION deklariert werden, sind in den Unter-
klassen der Klasse und in der Klasse selbst ansprechbar.

Privater Sichtbarkeitsbereich

Syntax

Diese Anweisung definiert den privaten Sichtbarkeitsbereich der Klasse
class. Alle Komponenten der Klasse, die im Bereich hinter der Anwei-
sung PRIVATE SECTION deklariert werden, sind nur in der Klasse selbst
ansprechbar.

7.2.1.2 Klassenkomponenten

Syntax

In den Sichtbarkeitsbereichen werden die Komponenten der Klassen defi-
niert. Folgende Deklarationsanweisungen sind für components möglich:

� TYPE-POOLS, TYPES, DATA, CLASS-DATA, CONSTANTS für Datentypen
und Datenobjekte (siehe Kapitel 6)

PUBLIC SECTION.

PROTECTED SECTION.

PRIVATE SECTION.

components.

� METHODS, CLASS-METHODS, EVENTS, CLASS-EVENTS für Methoden
und Ereignisse (siehe Abschnitt 7.4)

� INTERFACES zur Implementierung von Interfaces und ALIASES für
Alias-Namen von Interfacekomponenten (siehe Abschnitt 7.4)

Hinweis
Alle Komponenten einer Klasse liegen in einem Namensraum. Innerhalb
einer Klasse muss der Name einer Komponente unabhängig von seiner
Art (Datentyp, Attribut, Methode oder Ereignis) eindeutig sein. Die Kom-
ponenten eines implementierten Interfaces unterscheiden sich durch das
Präfix ifac~ (Name des Interfaces mit Interfacekomponenten-Selektor)
von den direkt deklarierten Komponenten der Klasse.

7.2.1.3 Globale Klassen

Syntax

Durch den Zusatz PUBLIC wird die Klasse class zu einer globalen Klasse
der Klassenbibliothek. Der Zusatz PUBLIC ist nur bei genau einer Klasse
eines Class-Pools möglich und wird beim Anlegen einer globalen Klasse
vom Class Builder erzeugt. Alle Klassen ohne den Zusatz PUBLIC sind
lokale Klassen ihres Programms.

Hinweis
Im öffentlichen Sichtbarkeitsbereich globaler Klassen dürfen keine eige-
nen Datentypen mit der Anweisung TYPES deklariert werden. Außerdem
können dort nur global bekannte Datentypen verwendet werden.

7.2.1.4 Definition von Unterklassen

Syntax

Durch den Zusatz INHERITING FROM wird die Klasse class durch Verer-
bung von der Oberklasse superclass abgeleitet und dadurch zu ihrer
direkten Unterklasse. Die Oberklasse superclass kann eine beliebige
nicht-finale, an dieser Stelle sichtbare Klasse sein. 

Jede Klasse kann nur eine direkte Oberklasse, aber mehrere direkte
Unterklassen haben (Einfachvererbung). Jede Klasse ohne den Zusatz
INHERITING FROM erbt implizit von der vordefinierten leeren Klasse

... PUBLIC ...

... INHERITING FROM superclass ...

object. Alle Klassen in ABAP Objects bilden einen Vererbungsbaum, in
dem es von jeder Klasse einen eindeutigen Pfad zum Wurzelknoten
object gibt.

Die Klasse class übernimmt alle Komponenten von superclass, ohne
deren Sichtbarkeitsbereich zu ändern. In der Unterklasse sind nur die
Komponenten des öffentlichen und geschützten Sichtbarkeitsbereichs der
Oberklasse sichtbar. Die Eigenschaften der übernommenen Komponen-
ten können nicht verändert werden. In einer Unterklasse können zusätz-
liche Komponenten deklariert und geerbte Methoden redefiniert, d.h.
ohne Änderung der Schnittstelle neu implementiert werden.

Hinweis
Die öffentlichen und geschützten Komponenten aller Klassen innerhalb
eines Pfads des Vererbungsbaums liegen im gleichen Namensraum. In
einer Unterklasse dürfen neue Komponenten nicht genauso heißen wie
öffentliche oder geschützte Komponenten, die von den Oberklassen
geerbt sind. 

7.2.1.5 Abstrakte Klassen

Syntax

Durch den Zusatz ABSTRACT wird eine abstrakte Klasse class definiert.
Von einer abstrakten Klasse können keine Instanzen erzeugt werden. Um
die Instanzkomponenten einer abstrakten Klasse zu verwenden, muss eine
nicht-abstrakte Unterklasse der Klasse instanziert werden.

7.2.1.6 Finale Klassen

Syntax

Durch den Zusatz FINAL wird eine finale Klasse class definiert. Von
einer finalen Klasse können keine Unterklassen abgeleitet werden. Alle
Methoden einer finalen Klasse sind implizit final und dürfen nicht explizit
als final deklariert werden.

Hinweis
In Klassen, die gleichzeitig abstrakt und final sind, sind nur die statischen
Komponenten verwendbar. Es können zwar Instanzkomponenten dekla-
riert werden, diese sind aber nicht verwendbar.

... ABSTRACT ...

... FINAL ...

Beispiel
Im folgenden Beispiel werden eine abstrakte Klasse c1 und eine finale
Klasse c2 definiert, wobei c2 von c1 erbt. c1 ist implizit Unterklasse der
leeren Klasse object. In c2 kann auf m1, aber nicht auf a1 zugegriffen
werden.

CLASS c1 DEFINITION ABSTRACT.
  PROTECTED SECTION.
    METHODS m1.
  PRIVATE SECTION.
    DATA a1 TYPE string 
            VALUE `Attribute A1 of class C1`.
ENDCLASS.

CLASS c2 DEFINITION INHERITING FROM c1 FINAL.
  PUBLIC SECTION.
    METHODS m2.
ENDCLASS.

CLASS c1 IMPLEMENTATION.
  METHOD m1.
    WRITE / a1.
  ENDMETHOD.
ENDCLASS.

CLASS c2 IMPLEMENTATION.
  METHOD m2.
    m1( ).
  ENDMETHOD.
ENDCLASS.

...

DATA oref TYPE REF TO c2.

CREATE OBJECT oref.
oref->m2( ).

7.2.1.7 Instanzierbarkeit

Syntax

... CREATE {PUBLIC|PROTECTED|PRIVATE} ...

Der Zusatz CREATE legt fest, in welchem Kontext die Klasse class instan-
ziert wird, d.h., wo die Anweisung CREATE OBJECT für diese Klasse aus-
geführt werden kann.

� Eine Klasse mit dem Zusatz CREATE PUBLIC kann überall dort instan-
ziert werden, wo die Klasse sichtbar ist. 

� Eine Klasse mit dem Zusatz CREATE PROTECTED kann nur in Metho-
den ihrer Unterklassen und der Klasse selbst instanziert werden.

� Eine Klasse mit dem Zusatz CREATE PRIVATE kann nur in Methoden
der Klasse selbst instanziert werden. Das bedeutet insbesondere, dass
sie auch nicht als vererbter Bestandteil von Unterklassen instanziert
werden kann.

Die Instanzierbarkeit einer Klasse hängt wie folgt von der direkten Ober-
klasse ab:

� Direkte Unterklassen von object oder einer Klasse mit dem Zusatz
CREATE PUBLIC erben implizit den Zusatz CREATE PUBLIC. Explizit
können alle CREATE-Zusätze angegeben werden.

� Direkte Unterklassen einer Klasse mit dem Zusatz CREATE PROTECTED
erben implizit den Zusatz CREATE PROTECTED. Explizit können alle
CREATE-Zusätze angegeben werden.

� Direkte Unterklassen einer Klasse mit dem Zusatz CREATE PRIVATE,
die keine Freunde der Klasse sind, erhalten implizit einen Zusatz
CREATE NONE. Sie sind nicht instanzierbar und es dürfen keine explizi-
ten CREATE-Zusätze angegeben werden. Direkte Unterklassen, die
Freunde der Klasse sind, erben implizit den Zusatz CREATE PRIVATE.
Explizit können bei Freunden privat instanzierbarer Oberklassen alle
CREATE-Zusätze angegeben werden.

Hinweis
Es empfiehlt sich, eine privat instanzierbare Klasse gleichzeitig final zu
machen, da ihre Unterklassen nicht instanziert werden können, wenn sie
keine Freunde der Klasse sind. 

7.2.1.8 Freunde

Syntax

Mit dem Zusatz FRIENDS macht die Klasse class die Klassen classi
bzw. die Interfaces ifaci zu ihren Freunden. Gleichzeitig werden sämtli-

... [GLOBAL] FRIENDS [classi] [ifaci].

che Unterklassen der Klassen classi, sämtliche Klassen, die eines der
Interfaces ifaci implementieren und sämtliche Interfaces, die eines der
Interfaces ifaci als Komponenteninterface enthalten, zu Freunden der
Klasse class. 

Die Freunde einer Klasse haben unbeschränkten Zugriff auf die geschütz-
ten und privaten Komponenten der Klasse und können uneingeschränkt
Instanzen der Klasse erzeugen. 

Die Freunde von class sind nicht automatisch auch Freunde der Unter-
klassen von class. Die Klasse class wird durch den Zusatz FRIENDS
nicht zum Freund ihrer Freunde.

Ohne die Zusätze GLOBAL oder LOCAL können für classi und ifaci alle
an dieser Stelle sichtbaren Klassen und Interfaces angegeben werden. Falls
globale Klassen und Interfaces der Klassenbibliothek zu Freunden
gemacht werden, ist zu beachten, dass in diesen die lokalen Klassen ande-
rer ABAP-Programme nicht sichtbar sind. Ein statischer Zugriff auf die
Komponenten der Klasse class ist in solchen Freunden nicht möglich.

Der Zusatz GLOBAL ist nur bei gleichzeitiger Verwendung des Zusatzes
PUBLIC, also für die globale Klasse eines Class-Pools erlaubt. Hinter
GLOBAL FRIENDS können andere globale Klassen und Interfaces der
Klassenbibliothek aufgeführt werden. Dieser Zusatz wird gegebenenfalls
beim Anlegen einer globalen Klasse vom Class Builder erzeugt. 

Beispiel
In folgendem Beispiel ist die Klasse c2 Freund des Interfaces i1 und
damit auch der implementierenden Klasse c1. Sie kann diese instanzieren
und auf ihre private Komponente a1 zugreifen. 

INTERFACE i1.
  ...
ENDINTERFACE.

CLASS c1 DEFINITION CREATE PRIVATE FRIENDS i1.
  PRIVATE SECTION.
    DATA a1(10) TYPE c VALUE 'Class 1'.
ENDCLASS.

CLASS c2 DEFINITION.
  PUBLIC SECTION.
    INTERFACES i1.
    METHODS m2.
ENDCLASS.

CLASS c2 IMPLEMENTATION.
  METHOD m2.
    DATA oref TYPE REF TO c1.
    CREATE OBJECT oref.
    WRITE  oref->a1.
  ENDMETHOD.
ENDCLASS.

Lokale Freunde globaler Klassen

Hierfür gibt es eine eigene Anweisung.

Syntax

Diese Anweisung macht die lokalen Klassen und Interfaces classi und
ifaci eines Class-Pools zu Freunden seiner globalen Klasse class. 

Diese Anweisung leitet keinen Deklarationsteil ein und darf nicht mit
ENDCLASS abgeschlossen werden. Sie muss im Class-Pool vor der Defini-
tion der lokalen Klassen und Interfaces classi und ifaci aufgeführt
werden, wenn diese Zugriff auf die privaten und geschützten Komponen-
ten der globalen Klasse class haben sollen.

7.2.2 Implementierungsteil

Implementierung der Methoden einer Klasse.

Syntax

Im Anweisungsblock CLASS class IMPLEMENTATION – ENDCLASS müs-
sen folgende Methoden einer Klasse class in beliebiger Reihenfolge im-
plementiert werden:

� alle nicht-abstrakten Methoden, die mit METHODS oder CLASS-
METHODS im Deklarationsteil der Klasse deklariert sind

� alle nicht-abstrakten Methoden von Interfaces, die mit der Anweisung
INTERFACES im Deklarationsteil der Klasse aufgeführt sind

CLASS class DEFINITION LOCAL FRIENDS [classi] [ifaci].

CLASS class IMPLEMENTATION.
  ...
  METHOD ...
    ...
  ENDMETHOD.
  ...
ENDCLASS.

� alle von Oberklassen geerbten Methoden, die im Deklarationsteil der
Klasse mit der Anweisung METHODS ... REDEFINITION aufgeführt
sind

Die Implementierung jeder Methode entspricht einem Verarbeitungsblock
METHOD – ENDMETHOD (siehe Abschnitt 4.2.1). Außerhalb von Methodenim-
plementierungen sind im Implementierungsteil keine Anweisungen er-
laubt. In einer Methodenimplementierung kann in Instanzmethoden auf
alle Komponenten und in statischen Methoden auf alle statischen Kompo-
nenten der eigenen Klasse zugegriffen werden. Für die Adressierung von
Komponenten der eigenen Klasse ist kein Komponentenselektor notwen-
dig. Innerhalb der Implementierung jeder Instanzmethode steht zur Laufzeit
eine implizit erzeugte lokale Referenzvariable namens me zur Verfügung, die
auf die aktuelle Instanz der Methode zeigt. 

Hinweis
Eine Klasse, die aufgrund ihres Deklarationsteils keine Methoden imple-
mentieren muss, hat entweder einen leeren oder gar keinen Implemen-
tierungsteil.

Beispiel
Im folgenden Beispiel müssen drei Methoden der Klasse c2 implemen-
tiert werden. Die Methode m1 in c1 ist abstrakt und muss nicht imple-
mentiert werden. 

INTERFACE i1.
  METHODS m1.
ENDINTERFACE.

CLASS c1 DEFINITION ABSTRACT.
  PROTECTED SECTION.
    METHODS m1 ABSTRACT.
ENDCLASS.

CLASS c2 DEFINITION INHERITING FROM c1.
  PUBLIC SECTION.
    INTERFACES i1.
    METHODS m2.
  PROTECTED SECTION.
    METHODS m1 REDEFINITION.
ENDCLASS.

CLASS c2 IMPLEMENTATION.
  METHOD m1.

    ...
  ENDMETHOD.
  METHOD m2.
    ...
  ENDMETHOD.
  METHOD i1~m1.
    ...
  ENDMETHOD.
ENDCLASS.

7.2.3 Klassen bekannt machen

In einigen Fällen muss die Definition einer Klasse vor ihrer Verwendung
explizit bekannt gemacht werden.

Syntax

Diese beiden Varianten der Anweisung CLASS dienen dazu, die Klasse
class unabhängig vom Ort der eigentlichen Definition im Programm
bekannt zu machen. Sie leiten keinen Deklarationsteil ein und dürfen
nicht mit ENDCLASS abgeschlossen werden. Die Zusätze haben folgende
Bedeutung:

� Die Variante mit dem Zusatz DEFERRED macht eine lokale Klasse vor
ihrer eigentlichen Definition im Programm bekannt. Das Programm
muss an späterer Stelle einen Deklarationsteil für class enthalten. Die
Anweisung ist notwendig, wenn man sich auf eine Klasse beziehen
will, bevor sie definiert wird. Ein Zugriff auf einzelne Komponenten ist
nicht vor der eigentlichen Definition möglich.

� Die Variante mit dem Zusatz LOAD lädt eine globale Klasse class aus
der Klassenbibliothek. Die Anweisung ist notwendig, wenn in einem
Programm auf eine der statischen Komponenten von class zugegriffen
oder ein Ereignisbehandler für class deklariert werden soll, bevor
class automatisch geladen wurde. Alle anderen Zugriffe laden class
automatisch.

Beispiel
Im folgenden Beispiel verwendet die Klasse c1 die Klasse c2 und umge-
kehrt. Deshalb muss eine der Klassen vor ihrer eigentlichen Definition
bekannt gemacht werden. Weiterhin wird die globale Klasse cl_gui_cfw
vor der Verwendung eines ihrer statischen Attribute explizit geladen. 

CLASS class DEFINITION { DEFERRED | LOAD }.

CLASS c1 DEFINITION DEFERRED.

CLASS c2 DEFINITION.
  PUBLIC SECTION.
    DATA c1ref TYPE REF TO c1.
ENDCLASS.

CLASS c1 DEFINITION.
  PUBLIC SECTION.
    DATA c2ref TYPE REF TO c2.
ENDCLASS.

CLASS cl_gui_cfw DEFINITION LOAD.

DATA state LIKE cl_gui_cfw=>system_state.

7.3 Definition von Interfaces

Die Definition eines Interfaces besteht aus einem Deklarationsteil, der
mit INTERFACE eingeleitet wird. Ein Interface hat im Gegensatz zu Klas-
sen keinen Implementierungsteil. Weitere Varianten von INTERFACE die-
nen dem Bekanntmachen von Interfaces in einem Programm.

7.3.1 Deklarationsteil

Definition eines Interfaces und Deklaration seiner Komponenten.

Syntax

Der Anweisungsblock INTERFACE – ENDINTERFACE definiert ein Inter-
face ifac. Zwischen INTERFACE und ENDINTERFACE werden die Kom-
ponenten components des Interfaces deklariert.

7.3.1.1 Interfacekomponenten

Syntax

INTERFACE

INTERFACE ifac.
  [components]
ENDINTERFACE.

components

Folgende Deklarationsanweisungen sind für components möglich:

� TYPE-POOLS, TYPES, DATA, CLASS-DATA, CONSTANTS für Datentypen
und Datenobjekte (siehe Kapitel 6)

� METHODS, CLASS-METHODS, EVENTS, CLASS-EVENTS für Methoden
und Ereignisse (siehe Abschnitt 7.4)

� INTERFACES zur Einbindung von Komponenteninterfaces und ALIASES
für Alias-Namen für deren Komponenten (siehe Abschnitt 7.4)

Hinweis
Alle Komponenten eines Interfaces liegen in einem Namensraum. Inner-
halb eines Interfaces muss der Name einer Komponente unabhängig von
seiner Art (Datentyp, Attribut, Methode oder Ereignis) eindeutig sein. Die
Komponenten eines eingebundenen Interfaces unterscheiden sich durch
das Präfix ifac~ (Name des Interfaces mit Interfacekomponenten-Selek-
tor) von den direkt deklarierten Komponenten.

Beispiel
Im folgenden Beispiel wird ein Interface i1 mit drei Interfacekomponen-
ten a1, m1 und e1 deklariert. Die Klasse c1 implementiert das Interface,
wodurch die Interfacekomponenten zu öffentlichen Komponenten der
Klasse werden, die über den Interfacekomponenten-Selektor (~) an-
sprechbar sind.

INTERFACE i1.
  DATA    a1 TYPE string.
  METHODS m1.
  EVENTS  e1 EXPORTING value(p1) TYPE string.
ENDINTERFACE.                    

CLASS c1 DEFINITION.
  PUBLIC SECTION.
    INTERFACES i1.
ENDCLASS.

CLASS c1 IMPLEMENTATION.
  METHOD i1~m1.
    RAISE EVENT i1~e1 EXPORTING p1 = i1~a1.
  ENDMETHOD.
ENDCLASS.

7.3.2 Interfaces bekannt machen

In einigen Fällen muss die Definition eines Interfaces vor seiner Verwen-
dung explizit bekannt gemacht werden.

Syntax

Diese beiden Varianten der Anweisung INTERFACE dienen dazu, das
Interface ifac unabhängig vom Ort der eigentlichen Definition im Pro-
gramm bekannt zu machen. Sie leiten keinen Deklarationsteil ein und
dürfen nicht mit ENDINTERFACE abgeschlossen werden. Die Bedeutung
der beiden Anweisungen für Interfaces ist identisch mit der Bedeutung
der Anweisungen CLASS in Abschnitt 7.2.3 für Klassen. 

7.4 Deklaration von Komponenten in Klassen 
und Interfaces

Im Deklarationsteil von Klassen und Interfaces werden die Komponenten
von Klassen und Interfaces deklariert, die wesentliche Komponenten
einer Klasse sind:

� Attribute

� Methoden

� Ereignisse

Attribute sind die Datenobjekte einer Klasse. Daneben ist auch die Dekla-
ration klasseneigener Datentypen möglich. Die Deklaration von Attribu-
ten und Datentypen erfolgt mit den allgemeinen Anweisungen TYPE-
POOLS, TYPES, DATA, CLASS-DATA und CONSTANTS, die mit Ausnahme
von CLASS-DATA auch in anderen Kontexten möglich und in Kapitel 6
beschrieben sind. 

Die Deklaration von Methoden und Ereignissen erfolgt mit speziellen
Anweisungen, die nur im Deklarationsteil von Klassen und Interfaces
möglich sind.

7.4.1 Methoden

Methoden werden mit den Anweisungen METHODS und CLASS-METHODS
deklariert. Sie bestimmen das Verhalten einer Klasse. Bei der Deklaration
wird die Schnittstelle einer Methode definiert. Für spezielle Aufgaben
gibt es unterschiedliche Arten von Methoden:

INTERFACE ifac { DEFERRED | LOAD }.

� allgemeine Methoden

� funktionale Methoden

� Konstruktoren

� Ereignisbehandler

7.4.1.1 Instanzmethoden

Die Anweisung METHODS deklariert Instanzmethoden. Instanzmethoden
sind an Objekte gebunden. Um Instanzmethoden zu verwenden, muss
zunächst ein Objekt der Klasse erzeugt werden. Instanzmethoden können
ohne Komponentenselektor auf alle Komponenten der eigenen Klasse
zugreifen. 

7.4.1.2 Allgemeine Instanzmethoden

Die allgemeinste Form der METHODS-Anweisung erlaubt die Definition
von Instanzmethoden mit beliebigen Ein- und Ausgabeparametern.

Syntax

Diese Anweisung deklariert eine allgemeine Instanzmethode meth. Mit
den Zusätzen ABSTRACT und FINAL wird die Methode abstrakt bzw. final
gemacht. 

Die übrigen Zusätze definieren die Parameterschnittstelle der Methode
und legen fest, welche Ausnahmen die Methode propagieren bzw. auslö-
sen kann. 

Parameterschnittstelle

Die Zusätze IMPORTING, EXPORTING und CHANGING definieren die Signa-
tur der Methode meth:

� IMPORTING definiert Eingabeparameter
Beim Aufruf der Methode muss für jeden nicht-optionalen Eingabepa-
rameter ein passender Aktualparameter angegeben werden. Der Inhalt

METHODS

METHODS meth [ABSTRACT|FINAL]
  [IMPORTING {parameteri} [PREFERRED PARAMETER p]]
  [EXPORTING {parameteri}]
  [CHANGING  {parameteri}]
  [{RAISING|EXCEPTIONS} {exci}].

des Aktualparameters wird beim Aufruf an den Eingabeparameter
übergeben.

� EXPORTING definiert Ausgabeparameter
Beim Aufruf der Methode kann für jeden Ausgabeparameter ein pas-
sender Aktualparameter angegeben werden. Der Inhalt des Ausgabe-
parameters wird bei fehlerfreier Beendigung der Methode an den
Aktualparameter übergeben.

� CHANGING definiert Ein-/Ausgabeparameter
Beim Aufruf der Methode muss für jeden nicht-optionalen Ein-/Ausga-
beparameter ein passender Aktualparameter angegeben werden. Der
Inhalt des Aktualparameters wird beim Aufruf an den Ein-/Ausgabepa-
rameter übergeben und bei Beendigung der Methode wird der Inhalt
des Ein-/Ausgabeparameters an den Aktualparameter übergeben.

Hinter jedem Zusatz werden die entsprechenden Formalparameter durch
die Angabe einer Liste parameteri definiert. 

Syntax von parameteri

Mit VALUE oder REFERENCE wird festgelegt, ob ein Parameter pi per
Wert oder per Referenz übergeben wird. Wenn nur ein Name pi angege-
ben wird, wird der Parameter standardmäßig per Referenz übergeben. Ein
per Referenz übergebener Eingabeparameter darf in der Methode nicht
verändert werden. 

Mit dem Zusatz typing muss jeder Formalparameter typisiert werden.
Die Syntax von typing ist in Abschnitt 8.2 beschrieben. Die Typisierung
eines Formalparameters bewirkt, dass bei der Übergabe eines Aktualpara-
meters dessen Datentyp gegen die Typisierung geprüft wird (siehe
Abschnitt 8.3). Weiterhin legt die Typisierung fest, an welchen Operan-
denpositionen der Formalparameter in der Methode verwendet werden
kann.

Mit OPTIONAL oder DEFAULT können Eingabeparameter und Ein-/Ausga-
beparameter als optionale Parameter definiert werden, wobei mit
DEFAULT ein Ersatzparameter defi angegeben werden kann. Für einen
optionalen Parameter muss beim Aufruf der Methode kein Aktualpara-
meter angegeben werden. Während ein Formalparameter mit dem
Zusatz OPTIONAL dann typgerecht initialisiert wird, übernimmt ein For-
malparameter mit dem Zusatz DEFAULT den Wert des Ersatzparameters

... { {VALUE(pi)}|[REFERENCE](pi)}|{pi} {typing}
      [OPTIONAL|{DEFAULT defi}] } ...

defi. Als Ersatzparameter defi kann jedes vom Typ passende Daten-
objekt angegeben werden, das an dieser Stelle sichtbar ist. 

Mit PREFERRED PARAMETER kann ein Eingabeparameter p der Liste para-
meteri hinter IMPORTING als bevorzugter Parameter gekennzeichnet wer-
den. Diese Angabe ist nur dann sinnvoll, wenn alle Eingabeparameter opti-
onal sind. Bei einem Aufruf der Methode mit der Syntax

[CALL METHOD] meth( a ).

wird dann der Aktualparameter a dem bevorzugten Parameter p zugeord-
net.

Hinweis
Falls ein Formalparameter als Referenzvariable typisiert ist und nicht in
der Prozedur verändert werden kann, wird die Typprüfung wie für einen
Narrowing Cast ausgeführt. Wenn er in der Prozedur verändert werden
kann, muss der Aktualparameter kompatibel zum Formalparameter sein.

Beispiel
Die Methode read_spfli_into_table des folgenden Beispiels hat
einen Eingabe- und einen Ausgabeparameter, die durch Bezug auf das
ABAP Dictionary vollständig typisiert sind.

CLASS flights DEFINITION.
  PUBLIC SECTION.
    METHODS read_spfli_into_table
       IMPORTING VALUE(id)  TYPE spfli-carrid
       EXPORTING flight_tab TYPE spfli_tab.
       ...
ENDCLASS.                  

Klassenbasierte Ausnahmen

Syntax

Mit dem Zusatz RAISING können klassenbasierte Ausnahmen exci an den
Aufrufer weitergereicht werden, die Unterklassen von CX_STATIC_CHECK
und CX_DYNAMIC_CHECK sind und in der Methode von der ABAP-Lauf-
zeitumgebung oder mit der Anweisung RAISE EXCEPTION ausgelöst
bzw. in dieses propagiert, aber nicht in einem TRY-Block behandelt wer-
den. 

... RAISING {exci} ...

Für exci können alle an dieser Stelle sichtbaren Ausnahmeklassen, die
Unterklassen obiger Oberklassen sind, angegeben werden. Die Ausnah-
meklassen müssen in aufsteigender Reihenfolge bezüglich ihrer Verer-
bungshierarchie angegeben werden.

Tritt in der Methode eine Ausnahme dieser Oberklassen auf, die weder
behandelt noch weitergereicht wird, führt dies entweder zu einem Syn-
taxfehler oder zu einer vom Aufrufer behandelbaren Ausnahme CX_SY_
NO_HANDLER�

Hinweis
In einer Methode, die mit dem Zusatz RAISING klassenbasierte Ausnah-
men propagiert, kann die Anweisung CATCH SYSTEM-EXCEPTIONS nicht
verwendet werden. Stattdessen sind die entsprechenden behandelbaren
Ausnahmen in einem TRY-Block abzufangen (siehe Kapitel 15).

Beispiel
In der folgenden Klasse math propagiert die Methode divide_1_by alle
Ausnahmen, die durch die Klasse CX_SY_ARITHMETIC_ERROR und
deren Unterklassen repräsentiert werden. Falls beispielsweise dem Einga-
beparameter operand beim Aufruf der Wert 0 übergeben wird, wird die
Ausnahme CX_SY_ZERODIVIDE ausgelöst, propagiert und kann – wie im
Beispiel gezeigt – vom Aufrufer in einem TRY-Block behandelt werden.

CLASS math DEFINITION.
  PUBLIC SECTION.
    METHODS divide_1_by
       IMPORTING operand TYPE I
       EXPORTING result  TYPE f
       RAISING   cx_sy_arithmetic_error.
ENDCLASS.

CLASS math IMPLEMENTATION.
  METHOD divide_1_by.
    result = 1 / operand.
  ENDMETHOD.
ENDCLASS.

...

DATA oref TYPE REF TO math.
DATA exc  TYPE REF TO cx_sy_arithmetic_error.

DATA result TYPE f.
DATA text   TYPE string.

CREATE OBJECT oref.
TRY.
    oref->divide_1_by( EXPORTING operand = ...
                       IMPORTING result = result ).
    text = result.
  CATCH cx_sy_arithmetic_error INTO exc.
    text = exc->get_text( ).
ENDTRY.
MESSAGE text TYPE 'I'.

Ausnahmen vor Release 6.10

Syntax

Mit dem Zusatz EXCEPTIONS wird eine Liste von Ausnahmen exci defi-
niert, die mit den Anweisungen RAISE oder MESSAGE RAISING in der
Methode ausgelöst werden können. Diese Ausnahmen sind – ähnlich wie
Formalparameter – an die Methode gebunden und können nicht propa-
giert werden. 

Wird eine solche Ausnahme in einer Methode ausgelöst und nicht mit
dem gleichnamigen Zusatz EXCEPTIONS der Anweisung CALL METHOD
beim Methodenaufruf behandelt, kommt es zu einem Laufzeitfehler. 

Hinweis
Die beiden Zusätze RAISING und EXCEPTIONS können nicht gleichzeitig
verwendet werden. Für Neuentwicklungen ab Release 6.10 wird empfoh-
len, mit klassenbasierten Ausnahmen zu arbeiten, die unabhängig von der
konkreten Methode sind.

Beispiel
In der folgenden Klasse math ist für die Methode divide_1_by eine
eigene Ausnahme arith_error definiert, die in der Methode mit der
Anweisung RAISE ausgelöst wird, wenn ein arithmetischer Fehler auftritt.
Falls beispielsweise dem Eingabeparameter operand beim Aufruf der
Wert 0 übergeben wird, wird die Ausnahme arith_error bei der
methodeninternen Behandlung der Ausnahme CX_SY_ZERODIVIDE aus-
gelöst und beim Aufruf der Methode behandelt.

CLASS math DEFINITION.
  PUBLIC SECTION.
    METHODS divide_1_by

... EXCEPTIONS {exci} ...

       IMPORTING  operand TYPE I
       EXPORTING  result  TYPE f
       EXCEPTIONS arith_error.
ENDCLASS.

CLASS math IMPLEMENTATION.
  METHOD divide_1_by.
    TRY.
        result = 1 / operand.
      CATCH cx_sy_arithmetic_error.
        RAISE arith_error.
    ENDTRY.
  ENDMETHOD.
ENDCLASS.

...

DATA result TYPE f.
DATA text TYPE string.

DATA oref TYPE REF TO math.
DATA exc  TYPE REF TO cx_sy_arithmetic_error.

CREATE OBJECT oref.
oref->divide_1_by( EXPORTING  operand  = ...
                   IMPORTING  result = result
                   EXCEPTIONS arith_error = 4 ).

IF sy-subrc = 0.
  WRITE result.
ELSE.
  WRITE 'Arithmetic error!'.
ENDIF.

Abstrakte Methoden

Syntax

Der Zusatz ABSTRACT ist nur in Klassen möglich, nicht in Interfaces.
Durch den Zusatz ABSTRACT wird eine abstrakte Methode meth definiert.
Eine abstrakte Methode wird nicht im Implementierungsteil ihrer Klasse

... ABSTRACT ...

implementiert und kann nur in abstrakten Klassen deklariert werden. Um
eine abstrakte Methode zu implementieren, muss sie in einer nicht-ab-
strakten Unterklasse mit dem Zusatz REDEFINITION redefiniert werden. 

Hinweis
Abstrakte Methoden sind zwar auch in Klassen definierbar, die sowohl
abstrakt als auch final sind, können aber nie implementiert werden und
sind daher nicht verwendbar.

Finale Methoden

Syntax

Der Zusatz FINAL ist nur in Klassen möglich, nicht in Interfaces. Durch
den Zusatz FINAL wird eine finale Methode meth definiert. Eine finale
Methode kann in einer Unterklasse nicht redefiniert werden. In finalen
Klassen sind alle Methoden automatisch final und der Zusatz FINAL ist
nicht erlaubt.

7.4.1.3 Funktionale Instanzmethoden

Funktionale Methoden haben beliebig viele Eingabeparameter und genau
einen Rückgabewert als Ausgabeparameter.

Syntax

Diese Anweisung deklariert eine funktionale Instanzmethode meth. Für
die Zusätze ABSTRACT, FINAL, IMPORTING, RAISING und EXCEPTIONS
gilt das Gleiche wie für allgemeine Instanzmethoden. 

Statt der Zusätze EXPORTING und CHANGING hat eine funktionale
Methode einen Zusatz RETURNING, der genau einen Formalparameter r
als Rückgabewert definiert. Der Rückgabewert muss mit VALUE per Wert
übergeben werden.

Für die Typisierung des Rückgabewerts r mit typing gelten die gleichen
Regeln wie für die Typisierung von IMPORTING-, EXPORTING- und CHAN-
GING-Parametern mit der Besonderheit, dass ein Rückgabewert immer

METHODS ... FINAL ...

METHODS meth [ABSTRACT|FINAL]
  [IMPORTING {parameteri}]
   RETURNING VALUE(r) {TYPE type_spec}|{LIKE dobj_spec}
  [{RAISING|EXCEPTIONS} {exci}].

vollständig typisiert sein muss: In typing kann keiner der generischen
Typen aus Tabelle 5.3 angegeben werden (siehe Abschnitt 8.2). 

Wenn funktionale Methoden wie in Abschnitt 2.2.5 beschrieben an den
Operandenpositionen von Anweisungen eingesetzt werden, an denen ein
Datenobjekt erwartet wird, bestimmt die vollständige Typisierung des
Rückgabewerts den Datentyp des Operanden. Bei der Ausführung einer
solchen Anweisung wird die funktionale Methode aufgerufen und der
Inhalt des Rückgabewerts ersetzt den Inhalt des Operanden.

Hinweise
� Wenn eine funktionale Methode in einer Operandenposition verwen-

det wird, können klassenbasierte Ausnahmen, die die Methode mit
RAISING propagiert, wie bei allgemeinen Methoden in einem TRY-
Block behandelt bzw. weiter propagiert werden. Die mit EXCEPTIONS
definierten Ausnahmen einer funktionalen Methode können dagegen
an Operandenpositionen nicht behandelt werden und führen immer
zu einem Laufzeitfehler. 

� Wenn eine funktionale Methode den gleichen Namen wie eine einge-
baute Methode hat, wird mit dem Ausdruck meth( a ) an einer Ope-
randenposition immer die funktionale Methode aufgerufen. Dieser Fall
kann nur in den Methoden der Klasse der funktionalen Methode auf-
treten.

Beispiel
Die funktionale Methode factorial des folgenden Beispiels hat einen
Rückgabewert fact vom Typ i. Die COMPUTE-Anweisung zeigt den Auf-
ruf der Methode an einer Operandenposition.

CLASS math DEFINITION.
  PUBLIC SECTION.
    METHODS factorial
       IMPORTING n TYPE i
       RETURNING value(fact) TYPE i.
ENDCLASS.

CLASS math IMPLEMENTATION.
  METHOD factorial.
    fact = 1.
    IF n = 0.
      RETURN.
    ELSE.
      DO n TIMES.

        fact = fact * sy-index.
      ENDDO.
    ENDIF.
  ENDMETHOD.
ENDCLASS.

...

DATA oref TYPE REF TO math.
DATA result TYPE i.

CREATE OBJECT oref.
COMPUTE result = oref->factorial( ... ).

7.4.1.4 Instanzkonstruktoren

Instanzkonstruktoren sind Methoden mit dem vorgegebenen Namen
constructor, die bei der Instanzierung ihrer Klasse automatisch aufgeru-
fen werden. Konstruktoren haben beliebig viele Eingabeparameter und
keine Ausgabeparameter.

Syntax

Diese Anweisung ist nur im öffentlichen Sichtbarkeitsbereich des Deklara-
tionsteils einer Klasse möglich. Sie deklariert den Instanzkonstruktor
constructor der Klasse. 

Jede Klasse hat in ihrem öffentlichen Sichtbarkeitsbereich eine vordefi-
nierte Methode namens constructor. Durch die explizite Deklaration
kann die Schnittstelle der Methode constructor klassenspezifisch defi-
niert und ihre Funktionalität implementiert werden. Ohne explizite
Deklaration übernimmt der Instanzkonstruktor die Signatur des Instanz-
konstruktors der direkten Oberklasse und ruft diesen implizit auf.

Falls der Instanzkonstruktor in einer Unterklasse implementiert wird,
muss der Instanzkonstruktor der Oberklasse explizit über die Pseudorefe-
renz super->constructor aufgerufen und seine Schnittstelle versorgt
werden. Ausgenommen hiervon sind nur die direkten Unterklassen des
Wurzelknotens object. Vor dem Aufruf des Oberklassenkonstruktors hat
ein Instanzkonstruktor nur auf die statischen Komponenten seiner Klasse

METHODS constructor [FINAL]
  [IMPORTING {parameteri}]
  [{RAISING|EXCEPTIONS} {exci}].

Zugriff. Nach dem Aufruf des Oberklassenkonstruktors kann auch auf
Instanzkomponenten zugegriffen werden.

Mit dem Zusatz IMPORTING können Eingabeparameter nach den gleichen
Regeln wie für allgemeine Methoden definiert werden. Die Zusätze
RAISING und EXCEPTIONS zur Propagierung bzw. Definition von Aus-
nahmen haben ebenfalls die gleiche Bedeutung wie bei allgemeinen
Methoden.

Der Instanzkonstruktor wird für jede Instanz einer Klasse genau einmal
durch die Anweisung CREATE OBJECT direkt nach ihrer Erzeugung aufge-
rufen. Dabei müssen allen nicht-optionalen Eingabeparametern passende
Aktualparameter zugeordnet werden und die Ausnahmen können behan-
delt bzw. weitergereicht werden. Ein Aufruf über CALL METHOD ist außer
beim Aufruf des Oberklassenkonstruktors über super->constructor
im redefinierten Konstruktor einer Unterklasse nicht möglich.

Instanzkonstruktoren sind implizit final. Der Zusatz FINAL kann zwar
angegeben werden, ist aber nicht notwendig. 

Hinweise

� Instanzkonstruktoren sind eine Ausnahme von der Regel, dass alle
öffentlichen Komponenten entlang eines Pfads im Vererbungsbaum in
einem Namensraum liegen. Der Instanzkonstruktor jeder Klasse hat
eine eigene Schnittstelle und eine eigene Implementierung. Ein
Instanzkonstruktor kann nicht redefiniert werden.

� Instanzkonstruktoren werden nur aus technischen Gründen im öffent-
lichen Sichtbarkeitsbereich einer Klasse deklariert. Die tatsächliche
Sichtbarkeit wird durch den Zusatz CREATE {PUBLIC|PROTECTED|
PRIVATE} der Anweisung CLASS DEFINITION gesteuert. 

Beispiel
Im folgenden Beispiel erbt die Klasse c2 von der Klasse c1. In beiden
Klassen ist der Instanzkonstruktor constructor explizit deklariert. Des-
halb muss er in beiden Klassen implementiert werden, wobei die Imple-
mentierung in c2 den Aufruf des Oberklassenkonstruktors enthalten
muss.

CLASS c1 DEFINITION.
  PUBLIC SECTION.
    METHODS constructor IMPORTING p1 TYPE any.
    ...
ENDCLASS.

CLASS c2 DEFINITION INHERITING FROM c1.
  PUBLIC SECTION.
    METHODS constructor IMPORTING p2 TYPE any.
    ...
ENDCLASS.

CLASS c1 IMPLEMENTATION.
  METHOD constructor.
    ...
  ENDMETHOD.
ENDCLASS.

CLASS c2 IMPLEMENTATION.
  METHOD constructor.
    ...
    super->constructor( p2 ).
    ...
  ENDMETHOD.
ENDCLASS.

7.4.1.5 Ereignisbehandler

Ereignisbehandler sind Methoden, die zwar auch mit CALL METHOD,
hauptsächlich aber durch das Auslösen eines Ereignisses einer Klasse oder
eines Interfaces aufgerufen werden können. Die einzig möglichen Formal-
parameter eines Ereignisbehandlers sind Eingabeparameter, die als Ausga-
beparameter des Ereignisses definiert wurden. 

Syntax

Diese Anweisung deklariert die Instanzmethode meth als Ereignisbehand-
ler für das Ereignis evt der Klasse class bzw. des Interfaces ifac. Für
class und ifac können alle an dieser Stelle sichtbaren Klassen und
Interfaces angegeben werden, die ein an dieser Stelle sichtbares Ereignis
evt als Komponente enthalten. 

Wenn das Ereignis evt ein Instanzereignis ist, kann der Ereignisbehandler
meth es in allen Objekten behandeln, deren Klassen gleich class oder
Unterklasse von class sind bzw. das Interface ifac direkt oder über eine
Oberklasse implementieren. Wenn das Ereignis ein statisches Ereignis ist,

METHODS meth [ABSTRACT|FINAL] 
   FOR EVENT evt OF {class|ifac}
  [IMPORTING {pi} [sender]].

kann es der Ereignisbehandler meth in der Klasse class und deren
Unterklassen bzw. in allen Klassen, die das Interface ifac implementie-
ren, behandeln.

Mit den Zusätzen ABSTRACT und FINAL können Ereignisbehandler genau
wie allgemeine Methoden entweder abstrakt oder final gemacht werden.

Der Zusatz IMPORTING definiert die Eingabeparameter des Ereignisbe-
handlers. Für pi können nur diejenigen Namen von Formalparametern
angegeben werden, die mit dem Zusatz EXPORTING der Anweisung
[CLASS-]EVENTS bei der Deklaration des Ereignisses evt in der Klasse
class bzw. dem Interface ifac als Ausgabeparameter des Ereignisses
definiert wurden. Die Zusätze TYPE bzw. LIKE und OPTIONAL bzw.
DEFAULT sind nicht möglich. Die Typisierung der Eingabeparameter, die
Eigenschaft, ob sie optional sind, und eventuelle Ersatzparameter werden
von der Deklaration des Ereignisses übernommen. Es müssen nicht alle
Ausgabeparameter des Ereignisses angegeben werden.

Falls evt ein Instanzereignis ist, kann zusätzlich zu seinen explizit definierten
Ausgabeparametern ein Formalparameter namens sender als Eingabepara-
meter eines Ereignisbehandlers definiert werden. Der Formalparameter
sender ist ein impliziter Ausgabeparameter jedes Instanzereignisses. Er ist
vollständig als Referenzvariable typisiert, die als statischen Typ die Klasse
class bzw. das Interfaces ifac hat, wie es in der Deklaration des Ereignis-
behandlers hinter EVENT evt OF angegeben ist. Wenn der Ereignisbehand-
ler durch ein Instanzereignis aufgerufen wird, bekommt er in sender eine
Referenz auf das auslösende Objekt übergeben.

Vor Release 6.10 wurde der statische Typ des Formalparameters sender
durch die Klasse bzw. das Interface bestimmt, in der bzw. dem das Ereig-
nis mit der Anweisung EVENTS deklariert wurde. Seit Release 6.10
bestimmt der Ereignisbehandler den Typ seines Formalparameters selbst. 

Beispiel
Die folgende Klasse picture enthält einen Ereignisbehandler handle_
double_click für das Instanzereignis picture_dblclick der globalen
Klasse cl_gui_picture. Der Ereignisbehandler übernimmt als Eingabe-
parameter zwei explizite Ausgabeparameter des Ereignisses sowie den
impliziten Parameter sender.

CLASS picture DEFINITION.
  PUBLIC SECTION.
    METHODS handle_double_click

            FOR EVENT picture_dblclick OF cl_gui_picture
            IMPORTING mouse_pos_x mouse_pos_y sender.
ENDCLASS.

CLASS picture IMPLEMENTATION.
  METHOD handle_double_click.
    ...
  ENDMETHOD.
ENDCLASS.

7.4.1.6 Instanzmethoden redefinieren

Eine in einer Oberklasse deklarierte Methode kann in einer Unterklasse
redefiniert werden, solange sie in der Oberklasse nicht als final gekenn-
zeichnet ist. Bei einer Redefinition wird die Schnittstelle der Methode
nicht geändert.

Syntax

Diese Anweisung ist nur in Unterklassen möglich und redefiniert eine
geerbte Instanzmethode meth. Sie bewirkt, dass die Methode meth im
Implementierungsteil der Unterklasse neu implementiert werden muss.
Die neue Implementierung verschattet in der aktuellen Klasse die Imple-
mentierung der Oberklasse. Die redefinierte Methode greift auf die priva-
ten Komponenten der redefinierenden Klasse zu und nicht auf eventuelle
gleichnamige private Komponenten der Oberklasse. In der redefinierten
Methode kann die Implementierung der direkten Oberklasse über die
Pseudoreferenz super->meth aufgerufen werden. 

Für meth kann außer dem Instanzkonstruktor jede nicht-finale Instanzme-
thode angegeben werden, die im öffentlichen oder geschützten Sichtbar-
keitsbereich einer Oberklasse der aktuellen Klasse deklariert ist. Insbeson-
dere kann meth eine abstrakte Methode einer abstrakten Oberklasse sein.
Die Redefinition muss im gleichen Sichtbarkeitsbereich wie die Deklara-
tion der Methode erfolgen. Die Schnittstelle und Art der Methode (allge-
meine oder funktionale Instanzmethode, Ereignisbehandler) wird bei
einer Redefinition nicht verändert.

Die Redefinition ist für die Unterklassen der redefinierenden Klasse gül-
tig, bis sie erneut redefiniert wird. Eine Methode ist entlang eines Pfads
im Vererbungsbaum so lange redefinierbar, bis sie final gemacht wird.

METHODS meth [FINAL] REDEFINITION.

Wenn der Zusatz FINAL bei der Redefinition verwendet wird, ist die
Methode ab der aktuellen Klasse final und in ihren Unterklassen nicht
mehr redefinierbar. 

Hinweis
Jede Objektreferenz, die auf ein Objekt einer Unterklasse zeigt, adressiert
unabhängig von ihrem statischen Typ die redefinierten Methoden. Dies
gilt insbesondere auch für die Selbstreferenz me->.

Beispiel
Im folgenden Beispiel wird die Methode m1 der Oberklasse c1 in der
Unterklasse c2 redefiniert, wobei die ursprüngliche Implementierung mit
super->m1 aufgerufen wird. Beide Methoden arbeiten mit dem privaten
Attribut a1 der jeweiligen Klasse. Beim Aufruf über die Referenzvariable
oref, die den statischen Typ c1 und den dynamischen Typ c2 hat, wird
die redefinierte Methode ausgeführt.

CLASS c1 DEFINITION.
  PUBLIC SECTION.
    METHODS m1 IMPORTING p1 TYPE string.
  PRIVATE SECTION.
    DATA a1 TYPE string VALUE `c1: `.
ENDCLASS.

CLASS c2 DEFINITION INHERITING FROM c1.
  PUBLIC SECTION.
    METHODS m1 REDEFINITION.
  PRIVATE SECTION.
    DATA a1 TYPE string VALUE `c2: `.
ENDCLASS.

CLASS c1 IMPLEMENTATION.
  METHOD m1.
    CONCATENATE a1 p1 INTO a1.
    WRITE / a1.
  ENDMETHOD.
ENDCLASS.

CLASS c2 IMPLEMENTATION.
  METHOD m1.
    super->m1( p1 ).
    CONCATENATE a1 p1 INTO a1.
    WRITE / a1.
  ENDMETHOD.
ENDCLASS.

...

DATA oref TYPE REF TO c1.

CREATE OBJECT oref TYPE c2.
oref->m1( `...` ).

7.4.1.7 Statische Methoden

Die Anweisung CLASS-METHODS deklariert statische Methoden. Statische
Methoden können über den Klassenkomponenten-Selektor (=>) unabhän-
gig von Objekten verwendet werden. Statische Methoden können ohne
Komponentenselektor nur auf die statischen Komponenten der eigenen
Klasse oder ihrer Oberklassen zugreifen. 

7.4.1.8 Allgemeine statische Methoden

Die allgemeinste Form der CLASS-METHODS-Anweisung erlaubt die Defi-
nition von statischen Methoden mit beliebig vielen Ein- und Ausgabepa-
rametern.

Syntax

Diese Anweisung deklariert eine allgemeine statische Methode meth. Mit
den Zusätzen wird die Parameterschnittstelle der Methode definiert und
festgelegt, welche Ausnahmen die Methode propagieren bzw. auslösen
kann. Die Syntax und Bedeutung der Zusätze ist dieselbe wie bei allge-
meinen Instanzmethoden. 

Hinweis
Statische Methoden sind nicht redefinierbar und können deshalb auch
nicht abstrakt oder final gemacht werden.

7.4.1.9 Funktionale statische Methoden

Funktionale Methoden haben beliebig viele Eingabeparameter und genau
einen Rückgabewert als Ausgabeparameter.

CLASS-METHODS

CLASS-METHODS meth
  [IMPORTING {parameteri} [PREFERRED PARAMETER p]]
  [EXPORTING {parameteri}]
  [CHANGING  {parameteri}]
  [{RAISING|EXCEPTIONS} {exci}].

kap07.fm  Seite 189  Donnerstag, 8. August 2002  2:43 14



190 Definition von Klassen und Interfaces

Syntax

Diese Anweisung deklariert eine funktionale statische Methode meth.
Die Zusätze haben genau dieselbe Syntax und Bedeutung wie bei funkti-
onalen Instanzmethoden.

Beispiel
Die folgende Klasse circle enthält zwei funktionale statische Methoden
circumference und area, die mit der Konstanten pi arbeiten. 

CLASS circle DEFINITION.
  PUBLIC SECTION.
    CONSTANTS pi TYPE f
                 VALUE '3.14159265358979324'.
    CLASS-METHODS: circumference IMPORTING r TYPE f
                                 RETURNING value(c) 

TYPE f,
                   area          IMPORTING r TYPE f
                                 RETURNING value(a) 

TYPE f.
ENDCLASS.

CLASS circle IMPLEMENTATION.
  METHOD circumference.
    c = 2 * pi * r.
  ENDMETHOD.
  METHOD area.
    a = pi * r ** 2.
  ENDMETHOD.
ENDCLASS.

...

DATA: circ TYPE f,
      area TYPE f,
      radius TYPE f.

radius = ...
circ = circle=>circumference( radius ).
area = circle=>area( radius ).

CLASS-METHODS meth
  [IMPORTING {parameteri}]
   RETURNING VALUE(r) {TYPE type_spec}|{LIKE dobj_spec}
  [{RAISING|EXCEPTIONS} {exci}].

7.4.1.10 Statische Konstruktoren

Statische Konstruktoren sind Methoden mit dem vorgegebenen Namen
class_constructor, die vor der ersten Verwendung ihrer Klasse auto-
matisch aufgerufen werden. Statische Konstruktoren haben keine Parame-
terschnittstelle.

Syntax

Diese Anweisung ist nur im öffentlichen Sichtbarkeitsbereich des Dekla-
rationsteils einer Klasse möglich. Sie deklariert den statischen Konstruktor
class_constructor der Klasse. 

Jede Klasse hat in ihrem öffentlichen Sichtbarkeitsbereich eine vordefi-
nierte Methode namens class_constructor. Durch die explizite
Deklaration kann ihre Funktionalität klassenspezifisch implementiert wer-
den. Ohne explizite Deklaration ist der statische Konstruktor leer. 

Der statische Konstruktor wird genau einmal pro Klasse und internem
Modus vor dem ersten Zugriff auf die Klasse automatisch ausgeführt. Ein
Zugriff auf die Klasse ist die Erzeugung einer Instanz der Klasse oder die
Adressierung einer statischen Komponente über den Klassenkomponen-
ten-Selektor. 

Beim ersten Zugriff auf eine Unterklasse wird im Vererbungsbaum nach
der nächsthöheren Oberklasse gesucht, deren statischer Konstruktor
noch nicht ausgeführt wurde. Erst im Anschluss daran wird der statische
Konstruktor dieser Oberklasse ausgeführt und danach sukzessive der aller
folgenden Unterklassen bis zur angesprochenen Unterklasse.

Wie alle statischen Methoden kann auch der statische Konstruktor nur
auf die statischen Komponenten seiner Klasse zugreifen. Weiterhin darf
der statische Konstruktor seine eigene Klasse nicht explizit ansprechen.

Hinweise

� Wie Instanzkonstruktoren sind auch statische Konstruktoren eine Aus-
nahme von der Regel, dass alle öffentlichen Komponenten entlang
eines Pfads im Vererbungsbaum in einem Namensraum liegen.

� Der Zeitpunkt der Ausführung des statischen Konstruktors liegt nicht
definitiv fest. Es wird allein seine Ausführung vor dem ersten Zugriff
auf die Klasse garantiert. In der Regel wird der statische Konstruktor
unmittelbar vor dem Zugriff auf die Klasse ausgeführt. Nur bei einem

CLASS-METHODS class_constructor.

Zugriff auf ein statisches Attribut der Klasse wird der statische Kon-
struktor schon zu Beginn des jeweiligen Verarbeitungsblocks aufgeru-
fen. Eine Programmierung, die sich auf die Ausführung des statischen
Konstruktors zu einem bestimmten Zeitpunkt verlässt, wird nicht emp-
fohlen.

Beispiel
Der statische Konstruktor der folgenden Klasse setzt beim ersten Zugriff
auf die Klasse das statische Attribut access_program mithilfe des Sys-
temfeldes sy-repid auf den Namen des Programms eines internen
Modus, das die Klasse als Erstes verwendet.

CLASS some_class DEFINITION.
  PUBLIC SECTION.
    CLASS-METHODS class_constructor.
  PRIVATE SECTION.
    CLASS-DATA access_program TYPE sy-repid.
ENDCLASS.

CLASS some_class IMPLEMENTATION.
  METHOD class_constructor.
    access_program = sy-repid.
  ENDMETHOD.
ENDCLASS.

7.4.1.11 Statische Ereignisbehandler

Statische Ereignisbehandler sind statische Methoden, die durch ein Ereig-
nis einer Klasse oder eines Interfaces aufgerufen werden. Die einzig mög-
lichen Formalparameter eines Ereignisbehandlers sind Eingabeparameter,
die als Ausgabeparameter des Ereignisses definiert wurden.

Diese Anweisung deklariert die statische Methode meth als Ereignisbe-
handler für das Ereignis evt der Klasse class bzw. des Interfaces ifac.
Syntax und Bedeutung der Zusätze sind identisch mit der Deklaration von
Instanzmethoden als Ereignisbehandler.

Statische Ereignisbehandler können unabhängig von einer Instanz der
Klasse von dem Ereignis evt aufgerufen werden.

CLASS-METHODS meth 
   FOR EVENT evt OF {class|ifac}
  [IMPORTING {pi} [sender]].

Beispiel
In der folgenden Klasse dialog_box wird ein statischer Ereignisbehandler
close_box für das Ereignis definiert, das ausgelöst wird, wenn ein Benut-
zer eine Dialogbox des GUI Control Framework (CFW) schließen will.

CLASS dialog_box DEFINITION.
  PUBLIC SECTION.
    METHODS constructor.
    ...
  PRIVATE SECTION.
    CLASS-DATA open_boxes TYPE i.
    CLASS-METHODS close_box
      FOR EVENT close OF cl_gui_dialogbox_container
      IMPORTING sender.
    ...
ENDCLASS.

CLASS dialog_box IMPLEMENTATION.
  METHOD constructor.
    ... " create a dialogbox
    open_boxes = open_boxes + 1.
  ENDMETHOD.
  METHOD close_box
    ... " close the dialogbox referred by sender
    open_boxes = open_boxes - 1.
  ENDMETHOD.
ENDCLASS.

7.4.2 Ereignisse

Ereignisse werden mit den Anweisungen EVENTS und CLASS-EVENTS
deklariert. Die Deklaration eines Ereignisses in einer Klasse bewirkt, dass
die Methoden der Klasse das Ereignis auslösen und dadurch die Ausfüh-
rung von Ereignisbehandlern bewirken können. Bei der Deklaration kön-
nen für ein Ereignis Ausgabeparameter definiert werden, für die beim
Auslösen Aktualparameter an die Ereignisbehandler übergeben werden.

7.4.2.1 Instanzereignisse

Die Anweisung EVENTS deklariert Instanzereignisse. Instanzereignisse sind
an Objekte gebunden. Sie können nur in Instanzmethoden der gleichen
Klasse ausgelöst werden. 

EVENTS

Syntax

Diese Anweisung deklariert ein Instanzereignis evt in einer Klasse oder
einem Interface. Das Instanzereignis evt kann in allen Instanzmethoden
der gleichen Klasse bzw. einer Klasse, die das Interface implementiert,
und – falls dort sichtbar – in den Instanzmethoden von Unterklassen mit
der Anweisung RAISE EVENT ausgelöst werden. 

Der Zusatz EXPORTING definiert die Parameterschnittstelle des Ereignis-
ses evt. Ein Ereignis hat ausschließlich Ausgabeparameter pi, die per
Wert übergeben werden. Die Syntax der Zusätze VALUE, OPTIONAL,
DEFAULT und die Typisierung mit typing entspricht der Definition der
Formalparameter in der Schnittstelle von Methoden mit der Anweisung
METHODS. 

Für alle nicht-optionalen Ausgabeparameter müssen, für alle optionalen
Ausgabeparameter können beim Auslösen des Ereignisses mit der Anwei-
sung RAISE EVENT passende Aktualparameter angegeben werden. Opti-
onale Parameter, für die kein Aktualparameter angegeben wird, werden
auf ihren typgerechten Initialwert bzw. auf den Ersatzparameter defi
gesetzt.

Bei der Deklaration eines Ereignisbehandlers mit dem Zusatz FOR EVENT
OF der Anweisung [CLASS-]METHODS können die Ausgabeparameter des
Ereignisses als Eingabeparameter des Ereignisbehandlers definiert wer-
den, wobei die Eigenschaften der Eingabeparameter von den in EVENTS
definierten Ausgabeparametern übernommen werden.

Neben den explizit mit EXPORTING definierten Ausgabeparametern hat
jedes Instanzereignis implizit einen Ausgabeparameter sender. Dieser
Ausgabeparameter ist vom Typ einer Referenzvariablen. Beim Auslösen
des Ereignisses mit der Anweisung RAISE EVENT bekommt sender impli-
zit die Referenz auf das auslösende Objekt zugewiesen. 

Bis Release 6.10 wurde der statische Typ des impliziten Parameters sen-
der durch die Klasse bzw. das Interface bestimmt, in der die Anweisung
EVENTS aufgeführt ist. Seit Release 6.10 wird für jeden Ereignisbehandler
der statische Typ des Eingabeparameters sender durch den Zusatz FOR
EVENT OF der Anweisung [CLASS-]METHODS bestimmt. 

EVENTS evt [EXPORTING { VALUE(pi) typing
                       [OPTIONAL|{DEFAULT defi}]  } ].

Hinweis
Der dynamische Typ des impliziten Formalparameters sender ist immer
die Klasse des Objekts, in dem das Ereignis ausgelöst wird. 

Beispiel
Im folgenden Interface window werden drei Ereignisse mit jeweils einem
expliziten nicht-optionalen Ausgabeparameter status deklariert. Die
Klasse dialog_window implementiert das Interface window. Das Inter-
face window_handler enthält Ereignisbehandler, die sowohl den explizi-
ten Parameter status als auch den impliziten Parameter sender impor-
tieren. Der statische Typ des Eingabeparameters sender ist die Klasse
dialog_window. Vor Release 6.10 war es das Interface window.

INTERFACE window.
  EVENTS: minimize EXPORTING VALUE(status) TYPE i,
          maximize EXPORTING VALUE(status) TYPE i,
          restore  EXPORTING VALUE(status) TYPE i.
ENDINTERFACE.

CLASS dialog_window DEFINITION.
  PUBLIC SECTION.
    INTERFACES window.
ENDCLASS. 

INTERFACE window_handler.
  METHODS: minimize_window
             FOR EVENT window~minimize OF dialog_window
             IMPORTING status sender,
           maximize_window
             FOR EVENT window~maximize OF dialog_window
             IMPORTING status sender,
           restore
             FOR EVENT window~restore OF dialog_window
             IMPORTING status sender.
ENDINTERFACE.

7.4.2.2 Statische Ereignisse

Die Anweisung CLASS-EVENTS deklariert statische Ereignisse. Statische
Ereignisse sind nicht an Objekte gebunden. Sie können in allen Methoden
der gleichen Klasse ausgelöst werden. 

CLASS-EVENTS

Syntax

Diese Anweisung deklariert ein statisches Ereignis evt in einer Klasse
oder einem Interface. Das statische Ereignis evt kann in allen Methoden
der gleichen Klasse bzw. einer Klasse, die das Interface implementiert,
und – falls dort sichtbar – in den Methoden von Unterklassen mit der
Anweisung RAISE EVENT ausgelöst werden. 

Der Zusatz EXPORTING definiert die Parameterschnittstelle des Ereignis-
ses evt. Syntax und Bedeutung der Zusätze VALUE, OPTIONAL, DEFAULT
und die Typisierung mit typing entsprechen der Definition von Instanz-
ereignissen mit der Anweisung EVENTS. 

Hinweis
Statische Ereignisse haben keinen impliziten Formalparameter sender. 

7.4.3 Interfaces implementieren bzw. einbinden

Interfaces können von Klassen implementiert oder von anderen Inter-
faces eingebunden werden. Für Interfacekomponenten können Aliasna-
men definiert werden.

Die Anweisung INTERFACES ist im öffentlichen Sichtbarkeitsbereich des
Deklarationsteils von Klassen und in Interface-Deklarationen möglich.

7.4.3.1 Interfaces in Klassen implementieren

Syntax 

Im öffentlichen Sichtbarkeitsbereich einer Klasse implementiert die Anwei-
sung INTERFACES das Interface ifac in der Klasse. Zudem ist die Angabe
von Zusätzen möglich, die die Eigenschaften von Interfacekomponenten
in der Klasse bestimmen.

CLASS-EVENTS evt [EXPORTING { VALUE(pi) typing
                       [OPTIONAL|{DEFAULT defi}]  } ].

INTERFACES

INTERFACES ifac 
  { { [ABSTRACT METHODS {methi}] 
      [FINAL METHODS {methj}] } 
    | [ALL METHODS {ABSTRACT|FINAL}] }
  [DATA VALUES {attri = vali}].

Für ifac können alle an dieser Stelle sichtbaren lokalen oder globalen
Interfaces angegeben werden. Durch die Implementierung werden die
Komponenten des Interfaces zu öffentlichen Komponenten der Klasse.
Eine Interfacekomponente namens comp hat in der Klasse den Namen
ifac~comp, wobei ifac der Name des Interfaces und das Zeichen ~ der
Interfacekomponenten-Selektor ist. Eine Klasse muss alle Methoden des
Interfaces in ihrem Implementierungsteil implementieren, solange sie sie
nicht als abstrakt erklärt.

Mit den Zusätzen ABSTRACT METHODS und FINAL METHODS können ein-
zelne Instanzmethoden methi des Interfaces in der implementierenden
Klasse abstrakt oder final gemacht werden. Dabei gelten die gleichen
Regeln wie für die Zusätze ABSTRACT und FINAL der Anweisung
METHODS. Insbesondere muss die gesamte Klasse abstrakt sein, wenn eine
Interface-Methode abstrakt gemacht wird und es darf keine Interface-
Methode gleichzeitig hinter ABSTRACT METHODS und FINAL METHODS
aufgeführt werden. Statt in der Klasse einzelne Interface-Methoden
abstrakt oder final zu machen, können mit dem Zusatz ALL METHODS
{ABSTRACT|FINAL} alle Interface-Methoden entweder abstrakt oder
final gemacht werden.

Mit dem Zusatz DATA VALUES können einzelnen Attributen attri
Anfangswerte zugeordnet werden. Der Zusatz erfüllt für Instanzattribute
die gleiche Funktion wie der Zusatz VALUE der Anweisung DATA für klas-
seneigene Attribute. Konstanten, die im Interface mit der Anweisung
CONSTANTS deklariert sind, können nicht hinter dem Zusatz DATA
VALUES aufgeführt werden.

Hinweise
Eine Klasse kann beliebig viele verschiedene Interfaces implementieren.
Alle von einer Klasse implementierten Interfaces liegen gleichberechtigt
auf einer Ebene. Wenn ein in einer Klasse implementiertes Interface ifac
zusammengesetzt ist, d.h. Komponenteninterfaces enthält, werden diese
unabhängig von ihrer Schachtelungshierarchie wie einzelne Interfaces in
der Klasse implementiert und ihre Komponenten werden nicht über den
Namen ifac, sondern den ihres Komponenteninterfaces angesprochen.
Die mehrfache Verwendung des Interfacekomponenten-Selektors in
einem Bezeichner (ifac1~ifac2~comp) ist prinzipiell nicht möglich.

Jedes Interface kommt in einer Klasse genau einmal vor und jede Inter-
facekomponente comp ist immer eindeutig über ifac~comp ansprechbar.
Auch die Komponenten eines Interfaces, das dadurch, dass es Interface-

komponente eines oder mehrerer anderer Interfaces ist, scheinbar mehr-
mals in einer Klasse implementiert werden kann, gibt es nur einmal.

7.4.3.2 Interfaces in Interfaces einbinden

Syntax 

In der Deklaration eines Interfaces bindet die Anweisung INTERFACES
das Interface ifac in dem deklarierten Interface ein. Die Angabe von
Zusätzen ist nicht möglich. Das Interface ifac wird dadurch zum Kompo-
nenteninterface eines zusammengesetzten Interface.

Ein Interface kann aus beliebig vielen verschiedenen Interfaces zusam-
mengesetzt werden. Alle Komponenteninterfaces liegen gleichberechtigt
auf einer Ebene. Wenn ein Komponenteninterface selbst zusammenge-
setzt ist, d.h. Komponenteninterfaces enthält, spielt die Schachtelungs-
hierarchie keine Rolle für die Zusammensetzung des Interfaces, aber für
die Zugriffsmöglichkeiten auf Interfacekomponenten. 

Um innerhalb eines zusammengesetzten Interfaces auf die Komponente
comp eines Komponenteninterfaces ifac zuzugreifen, kann der Ausdruck
ifac~comp mit dem Interfacekomponenten-Selektor (~) verwendet wer-
den. Die mehrfache Verwendung des Interfacekomponenten-Selektors in
einem Bezeichner (ifac1~ifac2~comp) ist prinzipiell nicht möglich. In
einem zusammengesetzten Interface kann mit dem Interfacekomponen-
ten-Selektor nur auf die Interfacekomponenten der Komponenteninter-
faces zugegriffen werden, die in diesem Interface mit der Anweisung
INTERFACES eingebunden werden. Um in einem zusammengesetzten
Interface auf Komponenten zuzugreifen, die nicht über die einmalige Ver-
wendung des Interfacekomponenten-Selektors adressierbar sind, müssen
mit der Anweisung ALIASES Aliasnamen für die Interfacekomponenten
deklariert werden.

Hinweis
Jedes Interface kommt mit seinen Komponenten in einem zusammenge-
setzten Interface genau einmal vor. Auch ein Interface, das dadurch, dass
es Interfacekomponente eines oder mehrerer anderer Interfaces ist,
scheinbar mehrmals in ein Interface eingebunden werden kann, gibt es
nur einmal.

Da es keine getrennten Namensräume für globale und lokale Interfaces
gibt, ist bei der Zusammensetzung lokaler Interfaces darauf zu achten,
dass es nicht zu Kombinationen gleichnamiger globaler und lokaler Inter-

INTERFACES ifac.

faces kommt, da diese bei ihrer Implementierung nicht auf einer Ebene
liegen könnten.

Beispiel
Das folgende Beispiel zeigt, wie mit der Anweisung INTERFACES Inter-
faces zusammengesetzt und implementiert werden. Die Klasse c1 imple-
mentiert die zusammengesetzten Interfaces i2 und i3. Obwohl i1 Kom-
ponenteninterface von i2 und i3 ist, ist es in der Klasse c1 nur einmal
vorhanden. Mit einer Referenzvariablen iref1 vom statischen Typ i1
wird ein Objekt der Klasse c1 erzeugt und die dort implementierte
Methode i1~m1 aufgerufen.

INTERFACE i1.
  METHODS m1.
ENDINTERFACE.

INTERFACE i2.
  INTERFACES i1.
  METHODS m2.
ENDINTERFACE.

INTERFACE i3.
  INTERFACES i1.
  METHODS m3.
ENDINTERFACE.

CLASS c1 DEFINITION.
  PUBLIC SECTION.
    INTERFACES: i2, i3.
ENDCLASS.

CLASS c1 IMPLEMENTATION.
  METHOD i1~m1.
    ...
  ENDMETHOD.
  METHOD i2~m2.
    ...
  ENDMETHOD.
  METHOD i3~m3.
    ...
  ENDMETHOD.
ENDCLASS.

...

DATA iref1 TYPE REF TO i1.

CREATE OBJECT iref1 TYPE c1.
iref1->m1( ).

7.4.3.3 Aliasnamen

Deklaration von Aliasnamen für Interfacekomponenten.

Syntax

Diese Anweisung deklariert im Deklarationsteil einer Klasse oder eines
Interfaces einen Aliasnamen alias für eine Komponente comp des Inter-
faces ifac. Das Interface ifac muss in der gleichen Klasse implementiert
bzw. im gleichen Interface eingebunden sein. Der Aliasname kann über-
all, wo er sichtbar ist, statt ifac~comp verwendet werden, um auf die
Interfacekomponente comp zuzugreifen.

Ein Aliasname gehört zu den Komponenten der Klasse und des Interfaces.
Er liegt im gleichen Namensraum wie die übrigen Komponenten und
wird an Unterklassen vererbt. In Klassen kann ein Aliasname in jedem
Sichtbarkeitsbereich deklariert werden. 

Hinweis
Bei der Implementierung von Interface-Methoden im Implementierungs-
teil von Klassen darf der Aliasname nicht verwendet werden. In der
Anweisung METHOD muss immer der vollständige Bezeichner ifac~meth
angegeben werden.

Beispiel
In den folgenden Interfaces i2, i3 und der Klasse c1 werden Aliasnamen
für die Methoden der eingebundenen bzw. implementierten Interfaces
deklariert. Im Implementierungsteil der Klasse werden die Interface-
Methoden in den METHODS-Anweisungen über den Interfacekomponen-
ten-Selektor implementiert. Innerhalb der Methoden können die Alias-
namen der Klasse verwendet werden. 

ALIASES

ALIASES alias FOR ifac~comp.

INTERFACE i1.
  METHODS meth.
ENDINTERFACE.

INTERFACE i2.
  INTERFACES i1.
  ALIASES m1 FOR i1~meth.
  METHODS meth.
ENDINTERFACE.

INTERFACE i3.
  INTERFACES i2.
  ALIASES: m1 FOR i2~m1,
           m2 FOR i2~meth.
  METHODS meth.
ENDINTERFACE.

CLASS c1 DEFINITION.
  PUBLIC SECTION.
    INTERFACES i3.
    ALIASES: m1 FOR i3~m1,
             m2 FOR i3~m2,
             m3 FOR i3~meth.
ENDCLASS.

CLASS c1 IMPLEMENTATION.
  METHOD i1~meth.
    ... m2( ) ...
  ENDMETHOD.
  METHOD i2~meth.
    ... m3( ) ...
  ENDMETHOD.
  METHOD i3~meth.
    ... m1( ) ....
  ENDMETHOD.
ENDCLASS.

